
Лицензия  

Авторские права на публикуемые материалы принадлежат автору книги Осипову 
Александру Викторовичу. Публикация данных материалов не предполагает извле-
чения какой-либо коммерческой выгоды. 

Публикуемые материалы защищены действующим законодательством об автор-
ском праве. Все предусмотренные этим законодательством права на опубликован-
ные материалы принадлежат их автору. 

Официальным источником для распространения материалов является Интернет-
сайт //pascalabc.net, ссылка на который при цитировании обязательна. Разрешает-
ся свободно копировать и распространять исключительно на безвозмездной осно-
ве опубликованные материалы при условии сохранения их в неизменном виде и с 
указанием авторства. Передача материалов третьим лицам разрешается при усло-
вии сохранения в них страницы с настоящей лицензией. Исключение делается для 
учебных заведений: при подготовке раздаточного материала допускается страни-
цу с лицензией не включать. Любые другие способы распространения опублико-
ванных материалов при отсутствии письменного разрешения автора запрещены. 

Запрещается любым организациям осуществлять любого рода лицензирование 
опубликованного материала и осуществлять какую бы то ни было иную связанную 
с авторскими правами деятельность без письменного разрешения автора. 

6 Символы и строки 

До сих пор речь шла о числах, изредка – о логическом типе данных (boolean). 
Несколько раз в тексте упоминались какие-то таинственные «строки» и «симво-
лы», которые будут рассмотрены в дальнейшем. Это время наступило. 

Текст любой (и этой тоже) книги набран при помощи букв, цифр и различных  
знаков, часть из которых даже не отображается (они называются служебными). В 
совокупности всё это составляет некий алфавит. Алфавит рассматривается как 
упорядоченный набор символов. 

Текст программы тоже состоит из символов, составляющих алфавит языка про-
граммирования. Символы складываются в более крупные конструкции, образуя 
строки символов. Чтобы отделить в тексте одну строку от другой, договаривают-
ся, какие символы будут служить разделителями строк. Обычно для этой цели 
выбираются один или два служебных символа. Исторически сложилось, что при-
знаком окончания строки являлся переход на новую строчку при печати. В зависи-
мости от типа печатающего устройства, символ завершения строки мог иметь  
шестнадцатеричный код $0D (1310) – «перевод строки», $0A (1010) – «возврат ка-
ретки», либо использовалась их последовательность $0D0A (и сейчас используется, 
к примеру, в Windows). 

За каждой клавишей на клавиатуре компьютера закреплены один или более сим-
волов. При нажатии клавиши в компьютер посылается соответствующий код и 
этому коду сопоставляется некоторый символ. (На самом деле клавиатура посыла-
ет не один код, а два – один при нажатии клавиши, второй – при ее отпускании. Это 
позволяет при зажатой клавише генерировать повтор символов.) В разных странах 
клавиатура одна и та же, но часть значков, нарисованных на ней, в каждой стране 
своя. Ничего удивительного, ведь алфавит любого языка всегда содержит какие-то, 
характерные для этого языка символы. Но часть символов никогда не меняется, за 
исключением очень малого количества языков, поэтому было решено таким сим-
волам присвоить  коды  от нуля до 127 и назвать их интернациональными сим-
волами. Остальные символы в каждом языке, имея одинаковые коды, могут раз-
личаться, образуя символы национальных алфавитов. 

Интернациональные символы и символы национальных алфавитов помещаются в 
так называемые кодовые таблицы. Если код символа однобайтный, в такой 
таблице можно закодировать 28 = 256 различных символов. Получается, что ровно 
половина кодов (0 .. 127) отведена для интернациональных символов, а другая 
половина (128 .. 255) – для символов национального алфавита. Говорят, что первая 
(или нижняя) половина у всех кодовых таблиц одинакова, а вторая (верхняя) 
отведена под локализацию. В данном случае локализация – это предоставление 
возможности операционной системе работать не на международном (английском) 

языке, а на языке конкретной (локальной) страны. Для доступа к символам того 
или иного национального алфавита нужно всего лишь сменить кодовую таблицу. 

В настоящее время однобайтные кодовые таблицы (KOI-8, ASCII) устарели. В самом 
деле, если требуется одновременно отображать символы, которые принадлежат 
разным кодовым таблицам, то нужно между этими таблицами постоянно пере-
ключаться. Но в универсальных языках программирования понятия «кодовая 
таблица» не существует. Приходится для выбора кодовой таблицы обращаться из 
программы к функциям операционной системы, что очень неудобно. Этот недоста-
ток устраняется при переходе на двухбайтные коды символов. 

Двухбайтный код позволяет адресоваться к 216=65536 различным символам. Этот 
код получил название Unicode («Юникод»). Но и тут есть разновидности. Суще-
ствуют смешанные кодировки, в которых интернациональные символы кодируют-
ся одним байтом, а прочие – двумя. Современные версии Windows работают с 
Unicode. 

Но давайте все же вернемся к PascalABC.NET. 

6.1 Символьный тип данных 
Данные символьного типа имеют тип char и занимают в памяти 2 байта. Исполь-
зуется кодировка Unicode. В тексте программы символьная константа (литерал) 
всегда заключается в одинарные кавычки. Если надо записать сам символ одинар-
ной кавычки, то кавычка удваивается: ' ' ' '.  

Нужно сразу уяснить, что в программе встречаются понятия как символа, так и его 
внутреннего кода. В базовом Паскале использовалась однобайтная кодировка, а 
кодовая таблица выбиралась в зависимости от локализации операционной систе-
мы. Так, Turbo/Borland Pascal и Free Pascal работали с кириллицей («русскими 
буквами») преимущественно в кодовой таблице CP866, Borland Delphi, работая в 
среде Windows, мог использовать таблицу CP1251. PascalABC.NET использует коди-
ровку Unicode. 

Новички в программировании часто смешивают понятия символа, обознача-
ющего цифру и имеющего тип char, и самой этой цифры, которая может быть 
неотрицательным однозначным числом одного из целочисленных типов. 
Важно понимать, что символ – это «картинка», рисунок, изображение числа, и 
за символом в программе кроется не значение изображенной цифры, а код 
этого символа, взятый из кодовой таблицы. 

Как и целочисленные типы, символьный тип относится к так называемому поряд-
ковому типу данных. Вероятно, тут имеет значение тот факт, что порядок следо-
вания символов друг за другом определяется порядком следования их внутренних 
кодов в кодовой таблице. Благодаря такой классификации в большинстве опера-
торов, методов и расширений PascalABC.NET на месте, где можно указать данное 

числового типа, разрешено указывать и символьные данные. Это сделает изучение 
данной части книги достаточно комфортным, потому что материал в основе уже 
знаком. 

Данные символьного типа, как и любого другого, описываются с указанием ключе-
вого слова var: 

var c1, p135, rz: char; // три переменные 

var Символ1: char; // одна переменная 

var s: sequence of char; // последовательность символов; 

var ar: array of char; // динамический массив символов 

Описание можно соединить с инициализацией: 

var a: char := 'a'; // тип указан явно 

var b:= 'b'; // автовыведение типа 

var kt:= ('A', 'B', 'C'); // кортеж из трех символов 

Далее по тексту этого раздела при описании конструкций языка как символы, 
так и переменные типа char будут обозначаться «c», если прямо не указано 
иное. 

Выводить символы можно любыми средствами, которые использовались для 
вывода других типов данных: Print и Println, Write и Writeln, а также при помощи 
расширений .Print и .Println. 

Получить десятичный код символа можно несколькими способами: 
 Ord(c) – код символа c в Unicode (тип word); 
 char.Code – то же самое, точечная нотация; 
 OrdAnsi(c) – десятичный код символа в однобайтной кодировке Windows (тип 

byte). 

Произвести обратную операцию – получить символ по его внутреннему коду, 
можно тоже несколькими способами: 
 Chr(word) – символ с указанным кодом Unicode; 
 #word – символ с указанным кодом Unicode; принимает только литерал; 
 ChrAnsi(byte) – символ с кодом в однобайтной кодировке Windows. 

Приведенная ниже программа выводит коды букв кириллицы из таблицы Unicode. 
Используется знание о том, что кириллица в этой таблице идет подряд, исключая 
буквы Ё и ё. Генератор Range строит последовательность из 64 символов кирилли-
цы (вот оно – первое проявление порядкового типа данных: Range работает и с 
символами). 

begin 

  Range('А', 'я').Select(c -> c.Code).Println 

end. 

1040 1041 1042 1043 1044 1045 1046 1047 1048 1049 1050 1051 1052 1053 1054 1055 

1056 1057 1058 1059 1060 1061 1062 1063 1064 1065 1066 1067 1068 1069 1070 1071 

1072 1073 1074 1075 1076 1077 1078 1079 1080 1081 1082 1083 1084 1085 1086 1087 

1088 1089 1090 1091 1092 1093 1094 1095 1096 1097 1098 1099 1100 1101 1102 1103 

Эту же задачу можно решить и более традиционно, с помощью цикла. Вспомним, 
что параметр цикла for может быть любого порядкового типа. Следовательно, и 
типа char. 

begin 

  for var c := 'А' to 'я' do 

    Print(Ord(c)) 

end. 

Соединяясь друг с другом, символы образуют строки (глава 6.2). Пока мы будем 
считать строки последовательностью символов. А еще лучше – динамическим 
массивом с индексами от единицы. Потому что, написав для строки s, например 
s[5], мы получим пятый от начала строки символ. Для соединения символов служат 
знаки операции «+» и «*». Знак + сцепляет символы между собой, знак * позволяет 
размножить символ нужное число раз путем умножения его на число. 

begin // p06001 

  Println((2 * ('м' + 'а') + ' ') * 5) 

end. 

мама мама мама мама мама 

А вот еще приятная неожиданность: если с символом сложить число, оно будет 
автоматически преобразовано к строке символов и эта строка сцепится с указан-
ным в выражении символом. 

begin 

  ('$' + 132.5).Println 

end. 

$132.5 

6.1.1 Анализ символов на принадлежность к группе 
Несмотря на то, что компьютеры возникли из-за потребности людей к счету, их 
работа во многом опирается на обработку символов. Мы нажали на клавиатуре 
клавишу с цифрой «5» – что получил компьютер? Правильно, числовой код нажа-
той клавиши. Особая программа – драйвер клавиатуры – обработала этот код и 
превратила его в другой – код символа ′5′ в кодовой таблице.  А где число 5? Его 
нет. Компьютер не получает набираемые с клавиатуры числа – он получает коды 
соответствующих символов. Чтобы начать считать, компьютер сначала должен 
сформировать числа, участвующие в счете. Но пока что он получает лишь коды 
символов – цифр и иных знаков, составляющих число . Поэтому дальше подключа-
ется еще одна программа, так называемый парсер, которая выделяет из кодов 
соседних символов те, которые могут дать изображение числа. Парсер преобразует 
коды в само число, в соответствии с его типом формирует внутреннее представле-
ние этого числа и только потом помещает в память на отведенное место. Вот 

сколько всего потребовалось написать, чтобы дать лишь примитивное представ-
ление о том, как компьютер обрабатывает нажатие клавиши. Получается, что без 
обработки символов нет и быть не может вычислений. 

Всегда ли нам нужно превращать символы в числа? Нет, лишь тогда, когда числа 
участвуют в вычислениях. Например, номер этого раздела 6.1.1 вовсе не предпола-
гает проведения над ним каких-то вычислительных манипуляций. Следовательно, 
имеются ситуации, когда символ должен оставаться символом. Например, пара 
символов «а» и «с», - это слово «ас» или число AC16? Тут все зависит от того, какой 
текст их окружает (говорят, что интерпретация последовательности символов 
зависит от окружения - контекста). 

Итак, понятно, что нужно уметь различать символы. Различает их, конечно же, 
программа. Которую, в свою очередь, пишет программист. Т.е. и вы в том числе. 
Различая символы, мы относим их к той или иной группе – буквы, цифры, знаки 
препинания, скобки, пробелы и переносы строк и т.п. А помогают нам в этом име-
ющиеся в языке функции, процедуры, методы и расширения. 

До сих пор мы не обращали особого внимания на понятие статических (классо-
вых) методов. Это потому, что время серьезно разобраться с классами пока не 
подошло. При вызове метода используется точечная нотация, в которой сначала 
указывается имя объекта, а потом, через точку, записывается вызываемый для 
объекта метод. Но в случае статических методов при вызове надо указывать не 
имя объекта, а класс (для символов - char). Имя объекта при этом записывается в 
круглых скобках, как обычный параметр. Поскольку ранее мы договорились, что 
char означает переменную или непосредственно символ, при упоминании классо-
вых методов будем писать с вместо char. Например, в формате записи 
char.IsLetter(c), char – это имя класса, а c означает переменную или непосред-
ственно символ. 

Если явно не указано иное, подразумевается, что коды символов имеют двух-
байтную кодировку (word) и принадлежат кодовой таблице Unicode 

6.1.1.1 Является ли символ буквой? 
 c.IsLetter – расширение возвращает True, если символ c принадлежит к группе 

букв и False в противном случае. 
 char.IsLetter(c) – статический метод класса, делающий то же самое. 

Буквами считаются латинские и кириллические буквы в обоих регистрах, а также 
буквы иных алфавитов, например, греческого. Именно эти буквы PascalABC.NET 
позволяет использовать в именах объектов программы. Давайте попытаемся 
оценить их количество. 

begin 

  Range(0, word.MaxValue).Where(n-> Chr(n).IsLetter).Count.Println // 48718 

end. 

Немало, оказывается! 48.7 тысяч. Жаль, большая часть из них – иероглифы. Но 
давайте разберемся, как работает эта «однострочная» программа. 

word.MaxValue возвращает максимальное значение, которое может поместиться в 
переменную типа word. Range строит последовательность от нуля до полученного 
значения, что позволяет получить все коды, теоретически помещающиеся в кодо-
вую таблицу. Последовательность фильтруется методом Where, использующим 
выражение Chr(n).IsLetter, которое строит символ с очередным кодом и проверяет, 
является ли он буквой. Метод .Count возвращает количество элементов последова-
тельности, прошедшее фильтр. 

6.1.1.2 Является ли символ цифрой? 
 c.IsDigit – расширение возвращает True, если символ c принадлежит к группе 

цифр (0, 1, … 9) и False в противном случае. 
 char.IsDigit(c) – статический метод класса, делающий то же самое. 

6.1.1.3 Является ли символ буквой или цифрой? 
char.IsLetterOrDigit(c) – статический метод класса возвращает True, если символ c 
принадлежит к группе букв или цифр и False в противном случае. 

6.1.1.4 Является ли символ пробельным? 
char.IsWhiteSpace(c) – статический метод класса возвращает True, если символ c 
принадлежит к группе пробельных символов и False в противном случае. 

Пробельные символы – термин, пришедший из типографской практики. Пробел – 
это пустое место, интервал между символами. Но приходилось ли вам когда-
нибудь задумываться, что длина пробела бывает различной? Пробельными в 
типографском и издательском деле называют непечатаемые (и неотображаемые) 
символы. Давайте посмотрим, какие коды им принадлежат. 

begin 

  Range(0, word.MaxValue).Where(n -> char.IsWhiteSpace(Chr(n))).Println 

end. 

9 10 11 12 13 32 133 160 5760 8192 8193 8194 8195 8196 8197 8198 8199 8200 8201 

8202 8232 8233 8239 8287 12288 

Символы с кодами 9 .. 13 относятся к так называемым управляющим символам – 
раньше они управляли внешними устройствами, имеющими печатающую каретку, 
например такими, как механический принтер. Символ с кодом 32 – это привычный 
пробел. Коду 160 соответствует – неразрывный пробел, запрещающий в его пози-
ции менять строку. Символы с кодом 8192 и последующие используются в про-
граммах верстки текстов – именно они обеспечивают пробелы различной длины. 

6.1.1.5 Является ли символ знаком препинания? 
char.IsPunctuation(c) – статический метод класса возвращает True, если символ c 
принадлежит к группе знаков пунктуации (разделителям) и False в противном 
случае. Познакомиться с этими символами нам также поможет программа. 

begin 

  Range(0, 2000).Select(n -> Chr(n)) 

      .Where(c -> char.IsPunctuation(c)).Println 

end. 

!"#%&'()*,-./:;?@[\]_{}¡«·»¿;·՚՛՜՝՞՟։֊ ٭۔٫٬؟٪؞؛؍،؉؉׳״׆־׀׃  ܍܌܋܊܉܈܇܆܅܄܃܂܁܀

Здесь просмотр ограничен символом с кодом 2000, чтобы не заполнять страницу 
знаками, которые в России никогда не используются и скорее всего, корректно не 
будут отображены при попытке их напечатать. Нам достаточно убедиться, что 
среди выведенных символов имеются все употребляемые знаки препинания. 

6.1.1.6 Принадлежит ли буква к верхнему регистру? 
 c.IsUpper – расширение возвращает True, если символ c принадлежит к буквен-

ным символам верхнего регистра (прописным) и False в противном случае. 
 char.IsUpper(c) – статический метод класса, делающий то же самое. 

В качестве маленькой премии отметьте, что если символ не является буквой, 
возвращается False, какому бы регистру он не принадлежал. Это позволяет не 
писать лишних условий вида if c.IsLetter then if c.IsUpper then … 

6.1.1.7 Принадлежит ли буква к нижнему регистру? 
 c.IsLower – расширение возвращает True, если символ c принадлежит к буквен-

ным символам нижнего регистра (строчным) и False в противном случае. 
 char.IsLower(c) – статический метод класса, делающий то же самое. 

Здесь также, если символ не является буквой, возвращается False, какому бы реги-
стру он не принадлежал. 

6.1.1.8 Располагается ли символ в указанном интервале? 
 c.InRange(c1, c2) – расширение возвращает True, если код символа c  принадле-

жит интервалу, в котором находятся коды символов, начиная от c1 и заканчи-
вая c2. В противном случае возвращается False. Например, буква L находится в 
интервале от 0 до R, но не находится в интервале от Д до Я. 

6.1.2 Операции преобразования символов 
К операциям преобразования символов отнесены: смена регистра буквенных 
символов, переход к следующим и предыдущим символам (в порядке следования 
их кодов в таблице), а также преобразование символа из группы «цифра» (подраз-
дел 6.1.1.2) к соответствующему этому символу однозначному числу. 

6.1.2.1 Приведение буквенного символа к верхнему регистру 
 c.ToUpper – расширение возвращает буквенный символ c, приведенный к верх-

нему регистру, если он принадлежит к нижнему регистру. В противном случае 
символ возвращается без изменения. 

 UpCase(c) – функция, делающая то же самое 
 UpperCase(c) – функция, синоним UpCase. 
 char.ToUpper(c) – статический метод класса, переводящий символ c на верхний 

регистр, если он принадлежит к нижнему регистру. В противном случае 
никаких действий не выполняется. 

6.1.2.2 Приведение буквенного символа к нижнему регистру 
 c.ToLower – расширение возвращает буквенный символ c, приведенный к ниж-

нему регистру, если он принадлежит к верхнему регистру. В противном случае 
символ возвращается без изменения. 

 LowCase(c) – функция, делающая то же самое 
 LowerCase(c) – функция, синоним LowCase. 
 char.ToLower(c) – статический метод класса, переводящий символ c на нижний 

регистр, если он принадлежит к верхнему регистру. В противном случае 
никаких действий не выполняется. 

6.1.2.3 Преобразование символа в число 
c.ToDigit – расширение возвращает буквенный символ c, преобразованный к изоб-
ражаемому им целому неотрицательному однозначному числу типа integer. 

Если символ не является цифрой, т.е. расширение .IsDigit возвращает для него 
False, при выполнении программы будет выдано сообщение «Ошибка времени 
выполнения: not a Digit» и программа завершится аварийно.  

6.1.2.4 Символ, предшествующий указанному 
 c.Pred – расширение возвращает буквенный символ, код которого предшествует 

коду символа c. 
 Pred(c) – функция, делающая то же самое.  

А какой код предшествует символу с кодом 0? Размышляя логически – никакой, 
потому что этим символом начинается кодовая таблица. Тогда что вернет вызов 
Chr(0).Pred? Увы, при выполнении будет сгенерировано исключение с сообщением 
«Ошибка времени выполнения: Значение было недопустимо малым или недопу-
стимо большим для знака». 

6.1.2.5 Символ, следующий за указанным 
 c.Succ – расширение возвращает буквенный символ, код которого следует за 

кодом символа c. 
 Succ(c) – функция, делающая то же самое.  

Какой код следует за символом с кодом 65535 – последним кодом в таблице симво-
лов? Дальнейшие рассуждения схожи с приведенными в предыдущем подразделе. 
Да и результат попыток получить код такого символа будет таким же. 

6.1.2.6 Смещение по кодовой таблице на указанное число сим-
волов 

 Dec(c) – процедура, заменяющая значение переменной, содержащей символ c на 
символ, предшествующий ему. Вспомните декремент – операцию Dec  - это она 
же, только для символов. 

 Dec(c, n) – процедура, заменяющая значение переменной, содержащей символ c 
символом, находящимся в кодовой таблице на n позиций раньше. Тоже 
декремент. 

 Inc(c) – процедура, заменяющая значение переменной, содержащей символ c на 
символ, следующий за ним. Это инкремент. 

 Inc(c, n) – процедура, заменяющая значение переменной, содержащей символ c 
символом, находящимся в кодовой таблице на n позиций дальше. И тоже это 
инкремент. 

6.1.3 Ввод символов 
В языке Паскаль ввод с клавиатуры всегда завершается нажатием клавиши Enter, а 
вводимые значения (кстати, состоящие из одного или более символов) разделяют-
ся пробелами или нажатием все той же клавиши Enter. Нажатие клавиши Enter в 
операционной системе Windows посылает комбинацию символов #13#10 («пере-
вод строки», «возврат каретки»), а в Unix-подобных системах – только #10. Но это 
тоже символы! Если не принять мер, они будут восприняты, как вводимые данные. 

Вводимые данные сначала сохраняются в некоторой области памяти (буфере), а 
затем анализируются. Каждый раз, когда нужна очередная порция данных, проис-
ходит обращение сначала к буферу, а если там данных недостаточно – тогда уже к 
клавиатуре. 

За прием данных с клавиатуры отвечают оператор Read(Элемент1, Элемент2, …) 
или его разновидность Readln(Элемент1, Элемент2, …). Readln (от английских слов 
Read Line – чтение строки) не воспринимает финальное нажатие клавиши Enter, 
как данные, более того, он полностью очищает буфер после ввода, даже если были 
прочитаны не все символы. 

При вводе числовых или логических данных наличие  или отсутствие служебных 
символов #13#10 в буфере непринципиально. Ведь в написании данных этих типов 
нет служебных символов, поэтому для парсера они лишь граница очередного 
введенного значения. Это позволяет использовать как Read, так и Readln. Но перед 
вводом символов буфер должен быть пуст, поэтому такому вводу должен предше-
ствовать ввод данных с помощью Readln. Это надо понять или хотя бы очень хоро-
шо запомнить. 

begin // p06002 

  var k := ReadInteger('Введите целое число:'); 

  Println('Вы ввели', k); 

  var x := ReadlnReal('Введите вещественное число:'); 

  Println('Вы ввели', x); 

  var c: char; 

  Print('Введите звездочку:'); 

  Readln(c); 

  Println('Вы ввели', c); 

  Println(15 * '-'); 

  k := ReadInteger('Введите целое число:'); 

  Println('Вы ввели', k); 

  x := ReadReal('Введите вещественное число:'); 

  Println('Вы ввели', x); 

  Print('Введите звездочку:'); 

  Readln(c); 

  Println('Вы ввели', c); 

  Writeln('*** Работа завершена ***') 

end. 

Введите целое число: 256 

Вы ввели 256  

Введите вещественное число: -24.78 

Вы ввели -24.78  

Введите звездочку: * 

Вы ввели *  

---------------  

Введите целое число: 34 

Вы ввели 34  

Введите вещественное число: 4.93 

Вы ввели 4.93  

Введите звездочку: Вы ввели  

  

*** Работа завершена *** 

Фигурной скобкой в приведенном диалоге, возникающем при работе с програм-
мой, отмечен «странный фрагмент». Конечно, эта странность заложена в програм-
му с целью продемонстрировать «А что будет, если… ?». И это место в программе 
тоже помечено фигурной скобкой. 

В первый раз звездочка была введена и успешно выведена. Во второй раз про-
грамма не делала паузы для ожидания ввода и не дала ничего ввести, а сразу выве-
ла пустую строку. Не пробел, нет, именно целую пустую строку. Как это могло 
случиться, где место ошибки? 

В первый раз вводу символа предшествовал оператор 

var x := ReadlnReal('Введите вещественное число:'); 

Во второй раз выполнялся оператор 

x := ReadReal('Введите вещественное число:'); 

Нашли различие? ReadlnReal и ReadReal. В первом случае использована разновид-
ность «ln», очищающая буфер ввода, во втором случае буфер не очищается перед 
вводом  символа. Именно поэтому во втором случае не делается обращение к 
клавиатуре, а символьное данное считывается из буфера. А там что? Там символ 
смены строки. Он и выводится: мы получили пустую строку. 

Ввод, предшествующий вводу символьных данных, должен обязательно вы-
полняться по команде с очисткой буфера (Readln, ReadlnInteger и им подобных). 
Вводимые символьные данные нельзя разделять пробелами или чем-либо 
еще: любое нажатие клавиши воспринимается, как символ.  

Имеются функции, осуществляющие ввод символьных данных с приглашением: 

ReadlnChar('ТекстПриглашения'); // ввод одного символа 

ReadlnChar2('ТекстПриглашения'); // ввод двух символов 

ReadlnChar3('ТекстПриглашения'); // ввод трех символов 

Приглашение может быть опущено и тогда используется формат ReadlnChar, 
ReadlnChar2, ReadlnChar3. 

Если очищать буфер ввода не нужно,   используются разновидности без «ln»: 
ReadChar, ReadChar2, ReadChar3; приглашение ко вводу также может присутство-
вать. 

Рассмотрим простой пример со вводом символов. 

begin 

  var(a, b, c) := ReadChar3('-->'); 

  Writeln(a, b, c, ' --> ', c, b, a) 

end. 

--> сон 

сон --> нос 

Здесь ввод используется лишь один раз, поэтому можно использовать как 
ReadChar3, так и или ReadlnChar3. В протоколе работы с программой видно, что все 
три символа вводились без пробелов. Для вывода использован оператор Writeln, 
чтобы символы не разделялись пробелами. Программа вывела принятые символы 
вначале в порядке их ввода, а затем в обратном порядке. 

Еще одна программа предлагает ввести буквы русского алфавита и проверяет 
правильность порядка их следования. Ее можно использовать в качестве неболь-
шого теста на знание алфавита. В программе используется порядок следования 
кириллических символов в кодовой таблице.  

begin // p06003 

  Println('Проверьте, насколько хорошо Вы знаете русский алфавит!'); 

  Println('Введите подряд русские буквы в алфавитном порядке.'); 

  Println('Регистр букв неважен; буква Ё учитывается'); 

  var mc := Range('А', 'Е') + 'Ё' + Range('Ж', 'Я'); 

  foreach var c in mc do 

   if ReadChar.ToUpper <> c then 

    begin 

      Println('Вы ошиблись!'); 

      exit // завершить работу программы 

    end; 

  Println('Молодец!') 

end. 

Проверьте, насколько хорошо Вы знаете русский алфавит!  

Введите подряд русские буквы в алфавитном порядке.  

Регистр букв неважен; буква Ё учитывается  

абвгдеёжзийклмнопрстуфхцчшщъыьэюя 

Молодец! 

Обратите внимание, что в теле цикла использован ввод посредством ReadChar, а не 
ReadlnChar. Программа запрашивает ввод символов подряд, но, как уже не раз 
писалось, анализ введенных данных (чтение буфера ввода) программа начнет 
выполнять только после нажатия клавиши Enter. К этому моменту в буфере долж-
ны накопиться 33 введенных символа с #13#10 в конце. В программе имеется цикл 
посимвольного ввода и анализа. Если использовать ReadlnChar, то после чтения 
первого символа буфер будет очищен и остальные введенные символы пропадут. 
Программа снова запросит ввод и так будет делать 32 раза (в русском алфавите 33 
буквы). Тот самый редкий случай, когда нужно именно Read, а не Readln. 

6.1.4 Пример: нахождение суммы в символьном виде 
Современные компьютеры имеют высокое быстродействие, 
но это не освобождает программиста от необходимости зна-
ния математики и умения думать. Рассмотрим пример задачи 
на вычисление суммы натуральных чисел, количество кото-

рых равно 10k. 

 Пусть нам требуется найти сумму для k=30. Если считать сумму «в лоб», последо-
вательно складывая натуральные числа от 1 до 1030, то современному компьютеру, 
выполняющему порядка 109 сложений за секунду, потребуется 1021 секунд, что 
примерно равно 31,7 триллионов лет. С учетом того, что оценочный возраст нашей 
Вселенной примерно в 2300 раз меньше, понятно, что бессмысленно затевать 
такой процесс. 

В математике числовая последовательность 1, 2, 3, … n называется арифметиче-
ской прогрессией. Не вдаваясь в подробности, используем формулу для расчета 
суммы членов данной прогрессии в нашей задаче. 

При k=30 мы будем иметь дело с числами порядка 1060, имеющими 60 цифр. По 
диапазону данных тут подойдет только BigInteger. Но пока мы с ними не умеем 
работать.  Точнее, не умеем задавать для них значения, превышающие величину 
int64.MaxValue. Поэтому идею с вычислениями суммы таким способом пока при-
дется отложить. 

Заметим, что величина 10k – это единица с k нулями, поэтому 10k+1 будет числом 
вида 1000…001, содержащим k-1 ноль. Число 5×10k-1 можно записать как пятерку  с 
k-1 нулем. Тогда S(k) будет иметь вид 5000…05000…00, где после пятерки следует 
k-1 ноль. Например, для k=1 получаем S(1)=55, для k=2 – S(k)=5050 и т.д. 

Вычисление свелось к выписыванию пятерок и нулей. А работать с символами мы 
уже умеем. Оформим решение функцией. 

function S(k: integer) := 2 * ('5' + '0' * (k - 1)); // p06004 

 

begin 

  S(30).Println 

end. 

Вот так выглядит наш результат: 

500000000000000000000000000000500000000000000000000000000000 

Сотые доли секунды вместо триллионов лет: стоило перед программированием 
задания немного подумать, не так ли? 

6.2 Строковый тип данных 
Как уже упоминалось, символы складываются в более крупные единицы текста – 
строки символов. В программировании под термином «строка» понимают совсем 
не то, что в других областях, например, в литературе или в издательском деле. 
Строка – это последовательность символов, принадлежащих некоторому алфавиту. 

В PascalABC.NET имеются два типа строк. Первый тип – строки, унаследованные от 
языка Turbo Pascal (в языке Паскаль, каким его описал Н.Вирт, строкового типа 
вообще не было). Длина таких строк ограничена 255 символами. В PascalABC.NET 
они именуются «короткими (размерными) строками» и относятся размерному 
типц. При описании короткой строки необходимо в квадратных скобках после 
ключевого слова string указывать ее максимальную длину – количество символов, 
которое строка может иметь в программе. Короткие строки оставлены в языке 

лишь для совместимости со старыми программами и для работы с типизирован-
ными файлами (часть 9). 

Второй тип строк – это современные строки, базирующиеся на строках Microsoft 
.NET Framework и их длина ограничена примерно двумя гигабайтами. При описа-
нии таких строк их длина не указывается. Это основной тип строк PascalABC.NET, 
они относятся к данным ссылочного типа, а их тип так и называется - «строки». 
Они, строго говоря, не являются строками .NET хотя бы уже потому, что их можно 
изменять и копировать прямым присваиванием, а нумерация символов в строках 
ведется от единицы. 

Каждый символ строки в памяти хранится в двух байтах таблицы Unicode. 

Строковые данные в PascalABC.NET описываются с ключевым словом string. Лите-
рал, представляющий строку, заключается в одинарные кавычки. Если внутри 
строки встречается символ одинарной кавычки, он удваивается. 

var st, МояСтрока, p18: string; // три строки 

var Строчка: string; // одна строка 

var sos: sequence of string; // последовательность строк 

var ar: array of string; // динамический массив строк 

var sh1: string[27]; // короткая строка, максимум 27 символов 

Описание строк можно соединять с инициализацией: 

var s1: string := 'Это строка'; // тип указан явно 

var s2 := '*** И это строка ***'; // автовыведение типа 

var kt := ('Это', 'тоже', 'строка'); // кортеж из трех строк 

Строка может состоять из одного символа, и даже может иметь и нулевую длину, 
т.е. не содержать ни одного символа. В последнем случае она записывается парой 
одиночных кавычек, следующих друг за другом. 

Если указано var d := ’R’, какого типа будет переменная d? Типа char, конечно же, – 
тут произойдет автовыведение типа. Если нужен тип string, в данной ситуации 
придется указать его явно или использовать явное приведение. 

var s1 := 'Я'; // символ типа char 

var s2: string := 'Я'; // строка типа string с длиной 1 – указан тип 

var s3 := string('Я'); // строка типа string с длиной 1 – явное приведение типа 

Короткие строки имеют размерный тип, а прочие cтроки – ссылочный, но при 
этом они могут вести себя, как объекты размерного типа. Строки можно рас-
сматривать как последовательности символов и как массивы символов, так что 
для них применима большая часть функций, методов и расширений, рассмот-
ренная в части 5. 

На первый взгляд, с обработкой строк в PascalABC.NET дела обстоят совсем непро-
сто. Имеется более ста различных процедур, функций, методов и расширений, 
предназначенных для работы со строками. Кажется невероятным, что можно 
запомнить, какие из них и когда следует использовать. Причина такого обилия 

кроется в требовании совместимости PascalABC.NET с базовым Паскалем. В Turbo 
Pascal (Borland Pascal) имелся некоторый набор процедур и функций для работы с 
короткими строками. В Borland Delphi появились и «длинные строки», принеся с 
собой большое количество добавочных функций, процедур и методов. Строки .NET 
Framework и новые парадигмы программирования PascalABC.NET существенно 
пополнили арсенал средств работы со строками. 

Рекомендуемые при написании новых программ функции, процедуры, методы 
и расширения для работы со строками ниже будут помечаться значком ⍟ 

К любому одиночному символу строки можно обратиться, указав номер его пози-
ции в строке, начиная от единицы, например a[5], str15[2 * i + 3]. Можно выполнить 
замену одиночного символа в строке при помощи оператора присваивания, напри-
мер, a[8] := ’Z’. 

6.2.1 Ввод строк 
Строка при вводе с клавиатуры складывается из отдельных символов, поэтому на 
нее распространяются особенности ввода символов, рассмотренные в подразделе 
6.1.3. Ввод строки завершается нажатием клавиши Enter, что не позволяет записать 
в нее символы смены строки #13 и/или #10 – их при необходимости нужно добав-
лять путем вставки в нужное место. Нажатие клавиши  пробела не разделяет вво-
димые строки, а вставляет в строку пробел, так что за один раз можно ввести лишь 
одну строку. Об этой особенности нужно помнить при вводе массива строк – ввод 
каждой строки завершается только нажатием клавиши Enter. То же самое касается 
ввода в одном операторе строк и других типов данных: строку следует завершать 
нажатием Enter, в то время как другие данные можно разделять и пробелами. 

Операторы Read и Readln осуществляют ввод, в том числе, строк, имена которых 
перечисляются в списке ввода. Использование оператора Read, не очищающего 
буфер ввода, требует большой осторожности, поэтому начинающим программи-
стам проще пользоваться оператором Readln и не смешивать в одном операторе 
ввод строк и прочих данных. Работа с нижеследующей программой должна дать 
ключ к пониманию такого совета. 

begin // p06005 

  var i1, i2, i3: integer; 

  var s1, s2: string; 

  Readln(i1, s1, i2, s2, i3); 

  Writeln(i1, '|', s1, '|', i2, '|', s2, '|', i3); 

end. 

Сеанс работы №1 (после ввода числа - пробел, после ввода строки – Enter): 

15 строка1 

-243 строка2 

16 

15| строка1|-243| строка2|16 

Пробелы, введенные после чисел, стали частью введенных строк. Это плохо. 

Сеанс работы №2 (после ввода любого данного – Enter): 

15 

строка1 

Ошибка времени выполнения: System.FormatException: Входная строка имела 

неверный формат. 

Непонятно, что произошло. Предполагалось, что будет считано число 15, затем 
строка «строка1», далее снова число… Раскрыть тайну поможет ввод, при котором 
строки также будут изображать числа, а чтобы их выделить, строки сделаем пяти-
символьными: 11111, 22222, 33333. 

Сеанс работы №3 (после ввода любого данного – Enter): 

15 

11111 

-243 

15||11111||-243 

«Тайна» генерации исключения в сеансе работы №2 раскрыта: ввод Enter делает 
строку пустой. 

Сеанс работы №4 (после ввода числа - пробел, после ввода строки – Enter): 

1 строка номер 1 

2 строка номер 2 

3 

1| строка номер 1|2| строка номер 2|3 

Можно вводить строки, содержащие, в том числе пробелы, но впоследствии пер-
вый символ каждой введенной строки нужно удалить (раздел 6.2.10). 

Может быть, использование оператора Read в этом случае будет более успешным? 
Удалим в тексте программы подстроку «ln» после «Read» и посмотрим результаты 
ее работы. 

Сеанс работы №1 (после ввода числа - пробел, после ввода строки – Enter): 

1 строка номер 1 

2 строка номер 2 

3 

1| строка номер 1|2| строка номер 2|3 

Результаты ничем не отличаются от полученных с использованием Readln. 

Сеанс работы №2 (после ввода любого данного – Enter): 

1 

11111 

2 

1||11111||2 

И в этом случае результаты аналогичны. Собственно, а чего другого следовало 
ожидать? Оператор Readln отличается от Read лишь тем, что опустошает буфер 
ввода после окончания своей работы. И только. 

Помимо операторов Read/Readln, строки можно вводить при помощи функций 
ReadlnString и ReadlnString(p), где p – строка приглашения ко вводу. Эти функции 
возвращают значение введенной строки и могут быть использованы в выражениях 
вместо имени строк, что сокращает текст программы. 

Существуют также разновидности функций ReadlnString2 и ReadlnString3, позво-
ляющие ввести две или три строки. 

Функции без «ln» - ReadSrting, ReadSrting2 и ReadSrting3 не очищают буфер ввода 
при окончании работы. 

Все упомянутые функции поддерживают необязательный параметр, задающий 
строку приглашения ко вводу. 

begin 

  var (s1, s2, s3) := ReadlnString3('Введите три строки через Enter:'); 

  ('(' + s1 + ')').Println; 

  ('(' + s2 + ')').Println; 

  ('(' + s3 + ')').Println; 

end. 

Такой ввод позволяет корректно задать содержимое всех строк: 

Введите три строки через Enter: первая строка 

строка №2 

третья строка 

(первая строка) 

(строка №2) 

(третья строка) 

Еще один пример, использующий результат ввода строки. 

begin // p06006 

  var s1 := ReadLnString('Привет, как тебя зовут?'); 

  var gr := ReadlnInteger(s1 + ', укажи свой год рождения:'); 

  case Sign(gr - 1971) of 

    -1: Println('Просто мамонт! О языке Паскаль тогда еще не знали...');  

    0: Println('Да вы с Паскалем ровесники!'); 

    1: Println('А язык Паскаль в это время уже существовал!') 

  end 

end. 

Ниже приведен сеанс работы с программой. 

Привет, как тебя зовут? Вася 

Вася, укажи свой год рождения: 1964 

Просто мамонт! О языке Паскаль тогда еще не знали... 

6.2.2 Вывод строк 
Выводить строки можно любыми средствами, которые использовались для вывода 
других типов данных: Print и Println, Write и Writeln, а также при помощи расшире-
ний .Print и .Println. 

В части расширений следует отметить, что они рассматривают строку, как после-
довательность отдельных символов, но выводят эти символы не через пробел, как 
для других типов данных, а через пустой символ. Тем не менее, расширения .Print и 
.Println могут принимать в качестве параметра строку-разделитель, что иллюстри-
рует следующая простая программа 

begin 

  'Тестовая строка'.Println; 

  'Тестовая строка'.Println(' '); 

  'Тестовая строка'.Println('*'); 

  'Тестовая строка'.Println(' = ') 

end. 

Рассмотрите вывод этой программы, чтобы понять, как работает вывод строк по 
расширениям .Print и .Println 

Тестовая строка 

Т е с т о в а я   с т р о к а 

Т*е*с*т*о*в*а*я* *с*т*р*о*к*а 

Т = е = с = т = о = в = а = я =   = с = т = р = о = к = а 

6.2.3 Длина строки 
Длина строки – это текущее количество символов в ней. 
 ⍟ s.Length – свойство, возвращает длину строки s; 
 s.Count – расширение, делает то же самое; 
 Length(s) – функция, делает то же самое. 

В приведенной ниже программе видно, что строки s2[6] и s3[20] ведут себя одина-
ково. 

begin // p06007 

  var s1 := 'Строка'; 

  var s2: string[6] := 'Строка'; 

  var s3: string[20] := 'Строка'; 

  Writeln('|',s1, '|', s2, '|', s3,'|'); // |Строка|Строка|Строка| 

  Println(s1.Length, s2.Length, s3.Length); // 6 6 6 

  Println(s1.Count, s2.Count, s3.Count); // 6 6 6 

  Println(Length(s1),Length(s2),Length(s3)); // 6 6 6 

end. 

Как получается, что описанная с длиной 20 символов строка, после инициализации 
литералом длиной 6 символов внешне выступает шестисимвольной? Понять 
происходящее позволит еще одна программа. 

begin // p06008 

  var s: string[20]; 

  s.Length.Println; // 0 

  s := 'Строка'; 

  s.Length.Println; // 6 

  s := 'Строка подлиннее'; 

  s.Length.Println; // 16  

  s := ''; 

  s.Length.Println; // 0 

  s := 'А это очень длинная строка'; // |А это очень длинная |  

  Writeln('|', s, '|'); 

  s.Length.Println // 20 

end. 

Что происходит? Короткая строка описана, как имеющая максимальную длину 20 
символов. Поскольку при первом определении длины строка не была инициализи-
рована, начальное значение присвоилось автоматически и строка оказалась пу-
стой. А длина пустой строки в символах равна нулю. Присваивание строке литера-
ла длиной 6 символов изменило длину строки до 6. При попытке поместить в s 
строку длиной 26 символов, вместилось только 20. Обратите внимание: никаких 
сообщений о потере части данных не выдается. Получается, что мы всегда выво-
дим фактическую длину строки. 

Строку можно усечь или удлинить, изменяя ее длину. Это можно сделать при 
помощи процедуры SetLength. Удлиняемая строка дополняется справа пробелами 
до нужной длины. 
⍟ SetLength(s, len) – усекает или удлиняет строку s до длины len. 

begin // p06009 

  var P: string->() := s -> Println('|' + s + '|'); // лямбда" для вывода 

  var s := 'Маша ела кашу'; 

  P(s); // '|' - чтобы видеть пробелы в начале и конце 

  SetLength(s, s.Length - 4); 

  P(s); 

  SetLength(s, s.Length + 6); 

  P(s) 

end. 

|Маша ела кашу|  

|Маша ела |  

|Маша ела       | 

Короткая строка с типом string[k] всегда имеет длину k+1 байт. В самом пер-
вом (левом) байте находится текущее количество символов, помещенных в 
строку. Поэтому Length и Count вычислениями не занимаются, а читают этот 
байт как переменную типа byte, но возвращают значение с типом integer. 

6.2.4 Арифметические операции со строками (+ и *) 
Операция сложения «+», примененная к строкам, соединяет их, «склеивает» в 
общую строку. Если же одним из операндов является значение числового выраже-

ния (за исключением типа BigInteger), а вторым – строка, то числовое значение 
предварительно преобразуется к строковому виду, а потом выполняется соедине-
ние. 

В операции умножения «*» один операнд должен быть строкой, второй - целочис-
ленным выражением. В результате строка соединяется сама с собой указанное 
значением выражения количество раз, давая новую строку. 

В одном операторе можно записывать и больше одного сложения и/или умноже-
ния. 

begin 

  Println('--- 25 + 3 * 8 = '+(25 + 3 * 8) + ' ---') 

end. 

Вывод программы будет выглядеть следующим образом: 

--- 25 + 3 * 8 = 49 --- 

Обязательно ли было использовать скобки в выражении (25 + 3 * 8) ? Да, обяза-
тельно. Если этого не сделать, то будут выполнены следующие шаги вычислений: 

'--- 25 + 3 * 8 = ' + 25 // получим '--- 25 + 3 * 8 = 25' 

3 * 8 // получим 24 

'--- 25 + 3 * 8 = 25' + 24 // получим '--- 25 + 3 * 8 = 2524' 

'--- 25 + 3 * 8 = 2524' + ' ---' // получим '--- 25 + 3 * 8 = 2524 ---' 

Не забывайте о скобках, когда будете «склеивать» выражения в строках. 

Можно «склеить» в строку и повторенный несколько раз отдельный символ, 
например, var s := 30 * '+'; 

StringOfChar(c, k) – функция, возвращающая строку, состоящую из повторенного k 
раз символа c. Вряд ли кто-то будет ее использовать, поскольку эквивалентная 
запись k * c (или c * k) и короче, и понятнее. 

6.2.5 Сравнение строк 
Содержимое строк, даже если их длина различна, можно сравнивать между собой. 
Строки сравниваются посимвольно слева направо до первого несовпадения кодов 
символов, либо до окончания одной или обеих строк. Для строк определены ре-
зультаты всех шести операций сравнения: <, <=, =, <>, >, >=. Большим считается тот 
символ, код которого больше. Если все символы более короткой строки совпали с 
символами более длинной, то большей строкой считается более длинная строка. 

begin 

  Println('Кошка' > 'Мышка'); // False, 'К' < 'М' 

  Println('12345' > '1234');  // True, начало совпадает, строка 1 длиннее 

  Println('ЛЕН' > 'ЛЁН'); // True, 'Ё' в Unicode идет перед 'А' 

  Println('лен' > 'лён'); // False, 'ё' в Unicode после 'я' 

  Println('Папа' <= 'мама'); // True, 'П' < 'м', вначале прописные буквы 

  Println('Пoдвох' = 'Подвох') // False, самая первая 'о' - латинская 

end. 

При сравнении строк возможны три варианта: первая строка длиннее, строки 
равны, вторая строка длиннее. Для анализа ситуации придется делать одно или 
два последовательных сравнения, что может порождать не отличающиеся боль-
шой наглядностью конструкции из вложенных «if». В PascalABC.NET включены 
средства, позволяющие частично преодолеть это неудобство. 
 CompareStr(s1, s2) – функция, выполняющее сравнение строк. Она возвращает 

значение типа integer. Когда первая строка длиннее, возвращается положитель-
ное, когда строки равны, возвращается 0 и когда первая строка короче, возвра-
щается отрицательное значение. Пытаться искать какую-то гарантированную 
закономерность в значениях этих чисел не следует. 

 ⍟ string.Compare(s1, s2) – статический метод, выполняющий сравнение строк. 
Он возвращает значение типа integer. Когда первая строка длиннее, возвраща-
ется 1, когда строки равны, возвращается 0 и когда первая строка короче, воз-
вращается -1. Получается своеобразная функция Sign( ). Метод удобно исполь-
зовать с оператором case для организации ветвления. 

 ⍟ string.Compare(s1, s2, IgnoreCase) – статический метод, делающий то же самое 
и при этом игнорирующий регистр  буквенных символов, если в качестве Ig-
noreCase указано True. Для запоминания не очень удачная мнемоника: запись 
string.Compare(s1, s2, True) ассоциируется с каким-то «настоящим», истинным 
сравнением, а тут наоборот – сравнение неточное, игнорирующее регистр. 

begin // p06010 

  var Эталон := 'Sqrt(A/(2*Pi*(B-Sqrt(B*B-C*C))))-Sin(D)'; 

  var a := Arr( 

      'Sqrt(A/(2*pi*(B-Sqrt(b*b-C*C))))-Sin(d)', 

      'Sqrt(A/(2*Pi*(B-Sqrt(B*B-С*C))))-Sin(D)', 

      'Sqrt(A/(2*Pi*(B-Sqrt(B*B-C*C)))-Sin(D))', 

      'Sqrt(A/(2*Pi*(B-Sqrt(B*B-C*C))))+Sin(D)'); 

  Println('Эталонная строка :', Эталон); 

  foreach var s in a do 

  begin 

    var r := string.Compare(Эталон, s, True); 

    case r of 

      -1: Println(s, ' - эталон меньше'); 

      0: Println(s, ' - отлично!'); 

      1: Println(s, ' - эталон больше') 

    end 

  end 

end. 

Будет получен следующий результат 

Эталонная строка : Sqrt(A/(2*Pi*(B-Sqrt(B*B-C*C))))-Sin(D)  

Sqrt(A/(2*pi*(B-Sqrt(b*b-C*C))))-Sin(d)  - отлично!  

Sqrt(A/(2*Pi*(B-Sqrt(B*B-С*C))))-Sin(D)  - эталон меньше  

Sqrt(A/(2*Pi*(B-Sqrt(B*B-C*C)))-Sin(D))  - эталон меньше  

Sqrt(A/(2*Pi*(B-Sqrt(B*B-C*C))))+Sin(D)  - эталон больше 

6.2.6 Копирование строк 
Cтроки копируются обыкновенным присваиванием вида s1 := s2. 

begin // p06011 

  var s1: string[20] := '12345'; 

  var s2: string[20] := 'abcde'; 

  Println(s1, s2); // 12345 abcde 

  s1 := s2; 

  Println(s1, s2); // abcde abcde 

  s2 := '67890'; 

  Println(s1, s2); // abcde 67890 

  s1 := 'абвгд'; 

  Println(s1, s2) // абвгд 67890 

end. 

Хорошо видно, что короткие строки независимы: изменения в одной строке никак 
не сказываются на другой. Но это короткие строки. А как с остальными? 

Строки имеют ссылочный тип и при выполнении присваивания s1 := s2 в s1 копи-
руется ссылка на строку s2. Но тогда любое изменение в одной из строк повлечет 
изменение в другой строке, а на практике этого не происходит. В этом легко убе-
диться, изменив описание строк в приведенной выше программе: 

  var s1 := '12345'; 

  var s2 := 'abcde'; 

Причина кроется в том, что PascalABC.NET при модификации строк использует 
высокоэффективный механизм «Копирование при записи», иногда называемый на 
профессиональном жаргоне «коровьим» (COW – Copy-on-Write). Его суть в том, что 
непосредственно перед модификацией создается копия строки, с которой выпол-
няется работа, а лишь затем ссылка обновляется. 

6.2.7 Выделение подстроки 
Подстрока – это часть строки s, полученная путем выборки некоторых ее символов, 
следующих подряд. Не забывайте, что символы в строке нумеруются от единицы. 
 ⍟ s.Left(k) – расширение, возвращает k левых символов строки; 
 LeftStr(s, k) – функция, делает то же самое;  
 ⍟ s.Right(k) – расширение, возвращает k правых символов строки; 
 RightStr(s, k) – функция, делает то же самое;  
 Copy(s, from, k) – функция, возвращает k символов, начиная с позиции from; 

 s.Substring(from, k) – метод, возвращает подстроку длиной k, выбирая из строки 
s символы, начиная с позиции from. Выход за пределы строки считается ошиб-
кой. Нумерация символов ведется от нуля. 

Для получения подстроки часто бывает удобным использовать срезы. Легче запо-
минается, короче запись.  

6.2.8 Срезы строк 
Со срезами вы уже знакомы по массивам (раздел 5.3.5). Срезы строк реализуются 
аналогично, но возвращают они подстроку, а не последовательность или массив 
отдельных символов. Срез может использоваться в выражениях везде, где может 
использоваться строка. 

begin // p06012 

  var s := 'Параграф'; 

  s[:5].Println; // Пара 

  s[5:].Println; // граф 

  s[2:5].Println; // ара 

  s[3::2].Println; // рга 

  s[8::-2].Println; // фраа 

end. 

Имеется также расширение .Slice, позволяющее получать срезы. Нумерация симво-
лов здесь ведется от нуля. 
 s.Slice(f, h) – возвращает срез строки s, начиная с позиции f и выбирая символы с 

шагом h; 
 s.Slice(f, h, k) – возвращает срез строки s длины не более k, начиная с позиции f и 

выбирая символы с шагом h; 

begin 

  var s := 'Параграф'; 

  s.Slice(0, 1, 4).Println; // Пара 

  s.Slice(4, 1).Println; // граф 

  s.Substring(0, 4).Println; // Пара 

  s.Substring(4).Println// граф 

end. 

Составим из букв слова «информатика» слово «карта». Возьмем слог «ка», затем 
букву «р» и буквы «ат» в обратном порядке (выделены в слове полужирным курси-
вом). Подобные задачи часто предлагают школьникам при изучении строк. 

begin // p06013 

  var s := 'информатика'; 

  var w := s.Right(2) + s[5] + s[8:6:-1]; 

  w.Println 

end. 

Здесь использованы расширение s.Left( ) (раздел 6.2.7), прямое обращение к сим-
волу строки по ее позиции s[ ] и срез s[ : : ]. Безусловно, это не единственное реше-
ние. 

6.2.9 Смена регистра символов 
Как и в случае с отдельными символами (раздел 6.1.2), в строке можно изменять 
регистр всех символов или их части. Смена выполняется функциями и расширени-
ями методов, поэтому исходная строка не меняется. 
 LowerCase(s) – возвращает строку s, приведенную к нижнему регистру; 
 ⍟ s.ToLower – то же самое; 
 UpperCase(s) – возвращает строку s, приведенную к верхнему регистру; 
 ⍟ s.ToUpper – то же самое. 

6.2.10 Удаление символов в начале и в конце строки 
Наиболее популярным при обработке строк является удаление пробелов перед 
первым непробельным символом (левые, или лидирующие пробелы) и после 
последнего непробельного символа(правые, или завершающие пробелы). 
 TrimLeft(s) – функция, возвращающая исходную строку s с удаленными лиди-

рующими пробелами; 
 s.TrimStart – метод, возвращающий исходную строку s с удаленными лидирую-

щими пробелами; 
 TrimRight(s) – функция, возвращающая исходную строку s с удаленными завер-

шающими пробелами; 
 s.TrimEnd – метод, возвращающий исходную строку s с удаленными завершаю-

щими пробелами; 
 Trim(s) – функция, возвращающая исходную строку s с удаленными лидирую-

щими и завершающими пробелами; 
 s.Trim – метод, возвращающий исходную строку s с удаленными лидирующими 

и завершающими пробелами. 

begin // p06014 

  var P: string->() := s -> Println('|' + s + '|'); // лямбда" для вывода 

  var s := '   Маша    ела    кашу    '; 

  P(s); // '|' - чтобы видеть пробелы в начале и конце 

  P(TrimLeft(s)); // |Маша    ела    кашу    | 

  P(s.TrimStart); 

  P(TrimRight(s)); // |   Маша    ела    кашу| 

  P(s.TrimEnd); 

  P(Trim(s)); // |Маша    ела    кашу| 

  P(s.Trim); 

end. 

Кроме пробела имеются и другие непечатаемые (пробельные) символы (глава 
6.1.1.4), поэтому встречаются задачи и на удаление всех таких символов. Функции 
TrimLeft( ) и TrimRight( ) эти символы не различают, а функцией Trim( ) и тремя 
перечисленными выше методами можно пользоваться. Если удаляется не более 5-6 
символов в начале или конце строки, предлагаемые методы работают быстрее 
функций, в противном случае функции оказываются проворнее. 

Что делать, если нужная строка окружена, например, ненужными точками, запя-
тыми и знаками «+» ? Можно ли удалить их? Описанные методы могут решить и 

это проблему. Для этого нужно указать в качестве аргумента удаляемые символы, 
разделяя их запятыми или использовать массив символов. Не забывайте, что в 
этом случае пробельные символы тоже должны быть указаны, если они могут 
быть нежелательными. 

Пример1: s.TrimStart(' ','.',',','+'); 

 

Пример2: var p := Arr(' ','.',',','+'); 

         s.TrimEnd(p); 

 

Пример3: var pa := Arr(' ','.',',', Chr(9), Chr(10), Chr(13)); 

         s.Trim(pa); 

6.2.11 Удаление подстрок 
 Удалить подстроку, например,  в позициях с 8 по 24, можно при помощи срезов. 

begin 

  var s := 'У Вани машина, у Наташи - мяч'; 

  s:=s[:8]+s[25:]; 

  s.Println // У Вани - мяч 

end. 

 Delete(s, from, k) – процедура, удаляющая из строки s подстроку длиной k сим-
волов, начиная с позиции from. Нумерация позиций ведется от нуля. 

begin 

  var s := 'У Вани машина, у Наташи - мяч'; 

  Delete(s, 7, 17); 

  s.Println // У Вани - мяч 

end. 

 ⍟ s.Remove(from) – расширение, удаляющее часть строки, начиная с позиции 
from и до конца. Нумерация позиций ведется от нуля. 

 ⍟ s.Remove(from, k) – расширение, удаляющее подстроку длины k, начиная с 
позиции from. Нумерация позиций ведется от нуля. Если удалить k символов 
невозможно, генерируется исключение. 

begin 

  var s := 'У Вани машина, у Наташи - мяч'; 

  s.Remove(7, 17).Println // У Вани - мяч 

end. 

 Усечь строку можно также при помощи процедуры SetLength (раздел 6.2.3). 
 Расширение .Remove умеет делать еще один вид удаления – по содержимому 

подстроки. Если в качестве параметра задать массив подстрок, либо перечис-
лить из через запятую, из строки будут удалены все вхождения каждой из под-
строк. 

 ⍟ s.Remove(ss) – удаление всех вхождений в строку s каждой из подстрок, 
заданных перечнем или массивом. 



196 | Часть 6   
 

begin // p06015 

  var s := 'Улишние Васловани машзасоряютина, у Насловаташи - мятекстч'; 

  s := s.Remove('лишние', 'слова', 'засоряют', 'текст'); // очистим текст 

  s.Println; // У Вани машина, у Наташи - мяч 

end. 

6.2.12 Инверсия 
Если в строке расположить все ее символы в обратном порядке, то полученную 
строку называют инвертированной относительно исходной или просто инверсией. 
Существуют также инверсии подстроки. PascalABC.NET предлагает несколько 
способов получения инверсии строки. 
 s.Inverse – расширение, возвращающее инверсию строки s; 
 ReverseString(s) – функция, возвращающее инверсию строки s; 
 ReverseString(s, from, k) – функция, возвращающая строку s, в которой инверти-

рована подстрока длиной k, начиная с позиции from. Нумерация позиций про-
изводится от нуля. 

Можно инвертировать строку с помощью среза и это самый короткий вариант 
записи инверсии. Строка может рассматриваться как последовательность симво-
лов, поэтому расширение .Reverse, также может выполнить инвертирование, но в 
результате будет получена не строка, а последовательность символов. Все эти 
варианты приведены в нижеследующем примере программы. 

begin // p06016 

  var s := 'А роза упала на лапу Азора'; 

  var s1 := s.Inverse; 

  Println(s1); // арозА упал ан алапу азор А 

  var s2 := s.Reverse; 

  Println(s2); // [а,р,о,з,А, ,у,п,а,л, ,а,н, ,а,л,а,п,у, ,а,з,о,р, ,А] 

  var s3 := ReverseString(s); 

  Println(s3); // арозА упал ан алапу азор А 

  var s4 := s[::-1]; // срез 

  Println(s4); // арозА упал ан алапу азор А 

  var s5 := ReverseString(s, 7, 13); 

  Println(s5); // А роза упал ан алапу Азора 

  var s6 := s[:8] + s[20:7:-1] + s[21:]; // подстрока с помощью срезов 

  Println(s6) // А роза упал ан алапу Азора 

end. 

6.2.13 Вставка подстроки 
Средств для вставки подстроки в строку всего два и функционально они несколько 
отличаются. Процедура Insert изменяет исходную строку, а метод Insert возвраща-
ет измененную строку в качестве результата. 

 Insert(ss, s, from) – процедура, вставляющая в строку s подстроку ss, начиная с 
позиции from. Указание несуществующей позиции приводит ко вставке под-
строки перед первым (при from <1) или после последнего символа строки; 

 s.Insert(from, ss) – метод, возвращающий строку, полученную путем вставки 
подстроки ss в исходную строку s с позиции from. Нумерация позиций ведется 
от нуля. Указание несуществующей позиции может вызвать исключение с со-
общением «Ошибка времени выполнения: Заданный аргумент находится вне 
диапазона допустимых значений». 

begin // p06017 

  var s := 'У меня есть конфета'; 

  s.Println; // У меня есть конфета 

  var s1 := s.Insert(12, 'вкусная '); 

  s1.Println; // У меня есть вкусная конфета 

  Insert('вкусная ', s, 13); 

  s.Println // У меня есть вкусная конфета 

end. 

6.2.14 Проверки в строке 
В процессе работы со строкой бывает нужно выполнить некоторые проверки ее 
содержимого. PascalABC.NET предоставляет достаточное количество средств для 
таких проверок. Результат проверки всегда имеет тип boolean и равен True в 
случае ее успешности. 
 StringIsEmpty(s, p) – функция. Проверяет, все ли символы строки s, начиная с 

позиции p, являются пробельными. О пробельных символах сказано в (6.1.1.4). 
Если все, значение p станет на единицу больше длины строки, в противном слу-
чае оно не измениться. В качестве p должна быть указана переменная целочис-
ленного типа; 

 s.StartWith(ss) – метод. Проверяет, начинается ли строка s с подстроки ss; 
 s.EndWith(ss) – метод. Проверяет, заканчивается ли строка s подстрокой ss; 
 s.Contains(ss) – метод. Проверяет, содержит ли строка s подстроку ss; 
 s.Between(s1, s2) – расширение. Проверяет, находится ли строка s между стро-

ками s1 и s2 (s1 ⩽ s ⩽ s2); 
 s.InRange(s1, s2) – то же самое; 
 s.IsMatch(reg, opt) – расширение (подраздел 6.2.19.2). Проверяет, удовлетворяет 

ли строка s регулярному выражению reg (раздел 6.2.19). С помощью opt можно 
задавать дополнительные опции. 

6.2.15 Разбиение строки на слова 
Операция разбиения строки на слова порождает массив строк, где каждый элемент 
является словом. Слово – это чаще всего последовательность непробельных сим-
волов, ограниченная не менее чем одним пробельным символом или концом стро-
ки. Безусловно, это не отвергает вариантов, когда слово должно удовлетворять 
более жестким требованиям, например, содержать только буквы, или только 
цифры, или не включать знаков препинания и т.д. 

 s.ToWords(cc) – расширение, возвращающее массив слов, полученных разбиени-
ем строки s; при этом в качестве разделителей слов могут использоваться один 
или более символов, перечисленных в сс через запятую, либо содержащихся в 
массиве символов сс. Пустые слова, порождаемые наличием нескольких следу-
ющих подряд символов из cc, в массив не попадают. Параметр сс можно не ука-
зывать, тогда разделителями слов считаются пробельные символы; 

 s.Split(cc) – метод, аналогичный расширению .ToWords. В отличие от него, в 
массив заносит и пустые строки. 

begin // p06018 

  var s := '   Карл   у   Клары украл    кораллы    '; 

  s.ToWords.Println('.'); // элементы массива выводятся через точку 

  s.Split.Println('.'); 

  s := '= =+Карл у  Клары+++украл = = =кораллы'; 

  s.ToWords('=', '+', ' ').Println('.'); 

  s.Split('=', '+', ' ').Println('.'); 

end. 

При выводе был использован символ «точка», позволяющий увидеть места, на 
которых находятся пустые строки. Каждая пара следующих подряд точек – это и 
есть место вывода пустой строки. 

Карл.у.Клары.украл.кораллы 

...Карл...у...Клары.украл....кораллы.... 

Карл.у.Клары.украл.кораллы 

....Карл.у..Клары...украл......кораллы 

Если список символов – разделителей слов большой, конструкция получается 
достаточно громоздкой. Ее можно сократить, определив массив символов раздели-
телей. 

var cc := '+-.,?!;:" '.ToCharArray; // метод рассмотрен в разделе 6.2.20 

s.Split(cc).Println(‘~’); 

Можно порекомендовать использовать расширение .ToWords, если слова разделе-
ны более чем одним символом и после обработки этих слов нет нужды сохранять 
точное количество разделителей. Для сохранения количества и положения разде-
лителей нужно использовать метод .Split. 

6.2.16 Замена подстроки в строке 
Достаточно часто бывает необходимость заменить в строке один контекст на 
другой. Например, поменять имя переменной в выражении на более осмысленное. 
 s.Replace(s1, s2) – метод, возвращающий строку, полученную из исходной стро-

ки s заменой всех подстрок s1 на подстроки s2. 

begin 

  var s := 'Sin(2*x-1)*Sqr(Cos(x+5))+0.4*x**3'; 

  s.Replace('x', 'УголНаклона').Println 

end. 

Sin(2*УголНаклона-1)*Sqr(Cos(УголНаклона+5))+0.4*УголНаклона**3 

Более сложные замены можно осуществить на основе расширения .RegexReplace, 
использующего регулярные выражения (подраздел 6.2.19.3). 

6.2.17 Сцепление (слияние) строк 
Слияние строк – операция, обратная разбиению строки. Несколько строк сливают-
ся в одну общую. При этом в месте слияния может находиться символ-разделитель, 
подстрока из нескольких разделителей или не находиться ничего. 
 Concat(s1, s2, …) – функция, возвращающая строку, которая является сцеплени-

ем (конкатенацией) строк s1, s2, … без использования разделителей; 
 string.Join(ss, del) – статический метод, возвращающий строку, полученную 

сцеплением подстрок, находящихся в массиве ss. Подстрока del используется в 
качестве разделителя; 

 ss.JoinIntoString(del) – расширение, возвращающее строку, полученную сцепле-
нием подстрок, находящихся в массиве или последовательности ss. Подстрока 
del используется в качестве разделителя; по умолчанию – символ пробела. 

Функция Concat особого интереса не представляет, поскольку она может быть 
заменена операцией сложения «+». А остальное имеет смысл рассмотреть. 

Результатом разбиения строки на слова (раздел 6.2.15) является динамический 
массив строк (array of string). Сделав в словах необходимые изменения, можно 
снова получить строку. Это дает возможность создавать короткие, пусть быть 
может, и не самые эффективные с точки зрения расходуемой памяти и времени 
выполнения программы. 

Традиционный путь обработки строки состоит в ее последовательном просмотре с 
целью выделения позиций, занимаемых очередным словом и некоторых действий 
над символами в этих позициях. Разбиение строки на слова и последующее сцеп-
ление обработанных строк позволяет легко написать программу обработки строк, 
при этом в минимальные сроки и без ошибок. 

В журнале «Информатика» №8 (561), 16-30.04.2008 имеется статья: Т. Богомолова, 
Ирина Фалина , В. Шухардина «Ступенька мастерства: решаем задачи на обработку 
строк». Под номером 18 для самостоятельного решения предлагается следующая 
задача: «Слова в строке разделяются пробелами. Зашифровать текст таким обра-
зом, чтобы каждое слово текста было записано в обратном порядке». Дается также 
и решение, записанное в базовом Паскале. Алгоритм предполагает использование 
метода барьерных элементов, который в данной реализации сводится к дописыва-
нию пробела в конец исходной строки. Ниже приводится авторский код програм-
мы. 
 
Операции изменения строк в PascalABC.NET выполняются весьма неэффективно и 
вызовы их в цикле существенно увеличивают время выполнения программы. 
Особенно ресурсоемки операции множественной вставки подстрок. Строки в Pas-
calABC.NET базируются на неизменяемых строках .NET Framework и любая моди-
фикация строки связана с операцией копирования. В определенной степени реша-

http://inf.1september.ru/view_article.php?ID=200800802
http://inf.1september.ru/view_article.php?ID=200800802

ет проблему использование класса StringBuilder, о котором имеет смысл подумать 
в случае, когда ваш алгоритм интенсивно модифицирует строки.  

var 

  s, s1: string; 

  t: char; 

  a, i, j, b, p: integer; 

 

begin 

  write('st ==> '); 

  readln(s); 

  s := s + ' '; {Добавили барьерный элемент} 

  i := 1; a := length(s); 

  repeat 

    if s[i] <> ' ' then 

    begin 

      s1 := ''; p := i; 

      while s[i] <> ' ' do 

      begin 

        s1 := s1 + s[i]; 

        i := i + 1; 

      end; 

      b := length(s1); 

      for j := 1 to b div 2 do 

      begin 

        t := s1[j]; s1[j] := s1[b - j + 1]; 

        s1[b - j + 1] := t 

      end; 

      delete(s, p, b); 

      insert(s1, s, p); 

    end 

    else i := i + 1 

  until i >= length(s); 

  writeln(s) 

end. 

32 строки текста на Паскале. Конечно, вы уже просмотрели код и поняли алгоритм. 
Нет? Хорошо, посмотрите еще пару минут. Снова не поняли? Ну что же, как гово-
рится, «тяжело в ученье – будет тяжело и в работе». Давайте теперь напишем то же 
самое на PascalABC.NET. 

begin // p06019 

  ReadlnString('st ==>').Split.Select(w -> w.Inverse) 

      .JoinIntoString.Println 

end. 

И это все? В один оператор??? Да, все. Кстати, эта программа в PascalABC.NET еще и 
работает вдвое быстрее – это к вопросу об «оптимальных по времени программах». 

Разберем наше решение. ReadlnString('st ==>') считывает с клавиатуры строку, 
выдав приглашение ко вводу, взятое из приведенной в журнале программы и 
возвращает ее в качестве результата. Дадим полученной строке условное имя s. 
Далее выполняется метод s.Split, разбивающий строку по пробельным символам 

(параметр не задан) и возвращающий массив полученных слов, в том числе пустых. 
Назовем этот массив a. Проецирование a.Select(w -> w.Inverse) применяет к каждо-
му элементу массива а расширение Inverse, в результате чего получается инверсия 
находящихся в массиве слов. Результат проецирования - последовательность, 
которую обозначим p. Расширение p.JoinIntoString объединяет элементы в строку, 
используя в качестве разделителя принятый по умолчанию пробел. Получаем 
некоторую строку s, которую s.Println и выводит на монитор. Писать все это доль-
ше, чем саму программу. 

6.2.18 Поиск в строке 
Под поиском понимается нахождение в строке номера позиции отдельного симво-
ла или подстроки. Такая позиция еще может называться индексом, если рассмат-
ривать строку как динамический массив символов; при этом индексирование 
производится от нуля. 
 Pos(ss, s) – функция. Возвращает позицию первого вхождения подстроки ss в 

строку s. Возвращает ноль, если подстрока не найдена; 
 PosEx(ss, s) – то же, для совместимости с Delphi; 
 Pos(ss, s, from) – функция. Возвращает позицию первого вхождения подстроки ss 

в строку s, начиная поиск с позиции номер from. Возвращает ноль, если под-
строка не найдена; 

 PosEx(ss, s, from) – то же, для совместимости c Delphi; 
 LastPos(ss, s) – функция. Возвращает позицию последнего вхождения подстроки 

ss в строку s. Возвращает ноль, если подстрока не найдена; 
 LastPos(ss, s, from) – функция. Возвращает позицию последнего вхождения 

подстроки ss в строку s, начиная поиск в обратном направлении с позиции from. 
Фактически, значение from ограничивает поиск первыми from символами. Воз-
вращает ноль, если подстрока не найдена; 

 s.IndexOf(ss) – метод. Возвращает позицию первого вхождения подстроки ss в 
строку s. Нумерация позиций от нуля. Возвращает -1, если подстрока не найде-
на; 

 s.IndexOf(ss, from, k) – метод. Возвращает позицию первого вхождения подстро-
ки ss в строку s, начиная поиск с позиции from и распространяя его на k симво-
лов. Можно представить себе, что сначала делается срез строки, а затем поиск 
ведется в нем, но индекс при этом возвращается относительно начала исходной 
строки. Нумерация позиций от нуля. Возвращает -1, если подстрока не найдена; 

 s.LastIndexOf(ss) – метод. Возвращает позицию полследнего вхождения под-
строки ss в строку s. Нумерация позиций от нуля. Возвращает -1, если подстрока 
не найдена; 

 s.LaseIndexOf(ss, from, k) – метод. Возвращает последнего первого вхождения 
подстроки ss в строку s, начиная поиск в обратном направлении с позиции from 
и распространяя его на k символов. Нумерация позиций от нуля. Возвращает -1, 
если подстрока не найдена; 

 s.IndexOfAny(cc) – метод. Возвращает индекс первого вхождения в строку s 
любого символа из массива cc. Нумерация позиций от нуля. Возвращает -1, если 
ни один символ не найден; 

 s.LastIndexOfAny(cc) – метод. Возвращает индекс последнего вхождения в строку 
s любого символа из массива cc. Нумерация позиций от нуля. Возвращает -1, ес-
ли ни один символ не найден; 

 s.MatchValue(reg, opt) – расширение. Возвращает первую подстроку в строке s, 
соответствующую регулярному выражению reg (глава 6.2.19.4). С помощью opt 
можно задавать дополнительные опции; 

 s.MatchValues(reg, opt) – расширение. Возвращает последовательность подстрок 
в строке s, соответствующих регулярному выражению reg (глава 6.2.19.5). С по-
мощью opt можно задавать дополнительные опции; 

 s.Matches(reg, opt) – расширение. Возвращает последовательность элементов 
типа Match из строки s, соответствующих регулярному выражению reg (глава 
6.2.19.6). С помощью opt можно задавать дополнительные опции. 

Найдем позиции всех вхождений слова в тексте при помощи функции Pos. Обрати-
те внимание на использование переменной i в левой и правой частях оператора 
присваивания с вызовом Pos. Еще одна особенность программы – «накопление» 
результатов в символьной строке r. 

begin // p06020 

  var s := 

      'Лениво дышит полдень мглистый,' + NewLine + 

      'Лениво катится река.' + NewLine + 

      'И в тверди пламенной и чистой' + NewLine + 

      'Лениво тают облака.' + NewLine + 

      '(Ф. И. Тютчев)'; 

  Writeln(s, NewLine, '-' * 45); 

  var (i, k, r) := (1, 0, 'Слово "Лениво" найдено в позициях '); 

  repeat 

    i := Pos('Лениво', s, i); 

    if i = 0 then break; 

    r += i + ' '; 

    i += 1; 

    k += 1; 

  until False; 

  if k = 0 then Println('Слово "Лениво" в строке не обнаружено') 

  else 

  begin 

    r.Println; 

    Println('Количество вхождений:', k) 

  end 

end. 

Программа находит позиции всех трех вхождений подстроки «Лениво». 

Лениво дышит полдень мглистый, 

Лениво катится река. 

И в тверди пламенной и чистой 

Лениво тают облака. 

(Ф. И. Тютчев) 

--------------------------------------------- 

Слово "Лениво" найдено в позициях 1 33 86  

Количество вхождений: 3 

Чтобы показать, насколько регулярные выражения упрощают работу со строками, 
сравните приведенную выше программу со следующей, дающей точно такой же 
результат. 

begin // p06021 

  var s := 

      'Лениво дышит полдень мглистый,' + NewLine + 

      'Лениво катится река.' + NewLine + 

      'И в тверди пламенной и чистой' + NewLine + 

      'Лениво тают облака.' + NewLine + 

      '(Ф. И. Тютчев)'; 

  Writeln(s, NewLine, '-' * 45); 

  var r := s.Matches('Лениво').Select(m -> m.Index + 1).JoinInToString; 

  if r.Length > 0 then 

  begin 

    Println('Слово "Лениво" найдено в позициях', r); 

    Println('Количество вхождений:', r.Where(c -> c = ' ').Count + 1) 

  end 

  else Println('Слово "Лениво" в строке не обнаружено') 

end. 

Каждый элемент Match хранит всю информацию относительно результатов поис-
ка, в том числе, саму найденную подстроку, ее длину и позицию в исходной строке. 
Это позволило спроектировать элементы на номера позиций в строке. Получилась 
последовательность типа integer, элементы которой расширение JoinIntoString 
преобразовало в строку, разделив значения пробелами. 

6.2.19 Использование регулярных выражений 
Термин регулярное выражение (далее по тексту РВ) пришел из теории формаль-
ных языков и отражает свойство математических выражений, называемое регу-
лярностью. Желающие разобраться в том, что такое регулярность, могут обра-
титься к специальной литературе. Официального стандарта, определяющего, что 
является РВ, а что нет, в мире не существует, поэтому имеется множество диалек-
тов РВ. 

Платформа Microsoft .NET Framework обеспечивает поддержку диалекта в стиле 
языка Perl, так что при изучении РВ можно пользоваться литературой и по этому 
языку. За поддержку РВ отвечает пакет System.Text.RegularExpressions, называе-
мый в обиходе RegEx. 

Внешне РВ представляет собой некоторую строку-шаблон, с которой сопоставля-
ется обрабатываемый текст. Результатом сопоставления будут подстроки, удовле-
творяющие РВ. На основании содержимого этих подстрок делается вывод о нали-
чии или отсутствии или чего-либо в тексте, осуществляется выборка подстрок или 
их замена. 

РВ состоит из метасимволов – специальных символьных комбинаций и литералов 
– всех прочих символов. Проводя аналогию с Паскалем,  метасимволы – это ключе-
вые слова, а литералы – все остальное. Метасимволы объясняют, как интерпрети-
ровать литералы в РВ для выполнения поиска в заданной строке. 

При работе с РВ учитывается набор значений параметров, определяющих специ-
фику поиска, например, нужно ли учитывать регистр символов, останавливаться 
после первого успешного поиска или продолжать его и т.д. Эти параметры могут 
задаваться как в самом РВ, так и в операторе языка, использующего это РВ. 

РВ посвящены целые книги. Понятно, что здесь невозможно сколь-нибудь подроб-
но рассказать о РВ или научить профессиональной работе с ними, поэтому будут 
приведены лишь базовые сведения, достаточные для понимания примеров ис-
пользования РВ. 

Неверно думать, что РВ способны решать любые проблемы обработки текстовых 
строк. Их удел – поиск (и, возможно, замена) подстрок. Есть отличная фраза: «Если 
для решения проблемы вы не к месту попытаетесь использовать регулярное вы-
ражение, у вас станет две проблемы. Вторая – как применить это регулярное вы-
ражение». 

Далее, когда речь будет идти о РВ, на основе которого производится поиск, наряду 
с термином «регулярное выражение» будет употребляться термин «шаблон». 

6.2.19.1 Некоторые метасимволы 
. (точка) – самый простой метасимвол. Он означает, что на данном месте в строке 
может встретиться любой символ. Например, слова клан, клен, клин и клон удо-
влетворяют шаблону ′кл.н′, а шаблону вида ′…′ будет удовлетворять любое 
трехбуквенное сочетание символов. При этом не следует забывать, что пробел 
тоже является символом, поэтому и он удовлетворяет метасимволу «точка». Если 
нужно сослаться на символ «точка», то нужно примерить экранирование – указать 
перед точкой символ «\», получая комбинацию «\.». 

\d – метасимвол «digit – цифра», означающий цифру от 0 до 9. Двухзначное число 
может быть изображено с помощью этого метасимвола в виде «\d\d». Если дана 
строка «У меня есть словарик на 23000 слов, изданный в 1974 году», то шаблон  
′\d\d′ позволит найти подстроки «23», «00», «19» и «74», а по ′\d\d\d′ – подстроки 
«230» и «197». 

\D – метасимвол, означающий не цифру. Вообще, в метасимволах РВ запись строч-
ных букв означает наличие соответствующих символов, а прописных – их отсут-
ствие. 

\w – метасимвол «word – слово», обозначающий все то, из чего строится слово, т.е. 
буквы, цифры и знак подчеркивания. Во многих операционных системах и языках 
программирования под буквами понимается только латинский алфавит. В Pas-
calABC.NET сюда включаются также буквы национальных алфавитов, поэтому не 
нужно беспокоиться по поводу кириллицы. Соответственно, все не \w обозначает-
ся, как \W. Во фразе «У меня есть словарик на 23000 слов, изданный в 1974 году» 
шаблону ′\w\w\w\w′ удовлетворяют подстроки «меня», «есть», «слов», «арик», 
«2300», «слов», «изда», «нный», «1974» и «году». 

\s – метасимвол «space – пробел», обозначающий любой пробельный символ. \S 
означает любой непробельный символ. 

^ – метасимвол, обозначающий начало строки, но только если он не входит в сим-
вольный класс (см. ниже). 

$ – метасимвол, обозначающий конец строки. 

Теперь вы можете потренироваться в нахождении подстрок. Для этого имеется 
достаточное количество онлайновых ресурсов. Один из наиболее подходящих для 
наших целей - //nregex.com 

Почему именно этот ресурс? Он работает с РВ, предназначенными для Microsoft 
.NET Framework. Ресурс имеет англоязычный интерфейс, но это непринципиально. 
При входе на Nregex в середине окна имеется большое поле с текстом «If I just had 
$5.00 …». Удалите этот текст и введите в этом поле строку, в которой будет произ-
водиться поиск, например, приведенную выше «У меня есть словарик на 23000 
слов, изданный в 1974 году». Шаблон вводится в строку под надписью «Regular 
Expression:», выполненную красным цветом, например, \d\d\d. Все найденные 
совпадения будут выделены. 

Онлайновое тестирование РВ – отличный инструмент для начинающих. Намного 
удобнее, чем видеть сообщение о неверном шаблоне после запуска своей програм-
мы. 

[…] – метасимвол, называемый «символьный класс». Внутри квадратных скобок 
перечисляются (без разделителей!) символы, которые могут представлять данный 
метасимвол, например [аи]. Шаблону кл[аи]н соответствуют подстроки «клан» и 
«клин». 

[^…] – метасимвол, называемый «инвертированный символьный класс». Ему соот-
ветствуют символы, не перечисленные внутри скобок. 

- (дефис) – метасимвол, но только в контексте символьного класса. Обозначает 
интервал. [A-Z] – все символы, коды которых находятся в интервале кодов от A до 
Z. Русская буква представляется в виде [А-Яа-яЁё]. Почему «Ёё» отдельно? Потому 
что в таблице Unicode эти две буквы стоят особняком. 

(…|...|…) – метасимвол «конструкция выбора». Выбирается одна из альтернатив, 
разделенных «|» - символом «ИЛИ». Например (кот|кошка). Инверсии для этой 
конструкции не существует. 

? - метасимвол, относящийся к квантификаторам (определяет количество по-
второв). Указывает, что символ, непосредственно предшествующий ему, является 
необязательным, т.е. может отсутствовать или присутствовать один раз. Напри-
мер, шаблону «сон.?» удовлетворяет и слово «сон», и слово «соня». 

+ - метасимвол, относящийся к квантификаторам. Указывает, что символ, непо-
средственно предшествующий ему, должен встретиться не менее одного раза. 
Шаблону «\d+» удовлетворяет последовательность цифр любой длины, а шаблон 
«[+-]?\d+» описывает произвольное целое число, которое может иметь знак. 

* - метасимвол, относящийся к квантификаторам. Указывает, что символ, непо-
средственно предшествующий ему, может отсутствовать или встретиться сколько 
угодно раз. Шаблону «[+-]?\d+\.\d\d*» соответствуют любые числа с фиксирован-
ной точкой. Разберем его состав. «[+-]?» указывает, что символ может присутство-
вать знак «+» или «-». «\d+» сообщает, что ожидается не менее одной цифры. «\.» - 
это знак точки, потому что указано экранирование. Без экранирования это обозна-
чало бы наличие одного произвольного символа. «\d\d*» - часть шаблона, указы-
вающая, что на этом месте должна быть одна цифра, за которой могут следовать 
другие цифры. Целиком шаблон читается примерно так: «Необязательный знак 
плюс или минус, за которым следует одна или более цифр, далее следует точка, за 
которой находится не менее одной цифры». Это адекватно описывает запись лю-
бого числа с фиксированной точкой. 

{m, n} – метасимвол, относящийся к квантификаторам. Указывает, что символ, 
непосредственно предшествующий ему, может встретиться не менее m и не более 
n раз. Если n=m, его можно опустить вместе с предшествующей запятой. Но если n 
не ограничено, его опускают, сохраняя запятую, и квантификатор {m,} означает «не 
менее m раз». Шаблон «\d{3}» определяет натуральное трехзначное число, а шаб-
лон «\d{1,2}\.\d{1,2}.\d{4}» - дату в формате дд.мм.гггг. 

6.2.19.2 IsMatch – наличие подстроки в строке 
Расширение s.IsMatch(reg, opt) проверяет, удовлетворяет ли строка s регулярному 
выражению reg. С помощью opt можно задавать дополнительные опции. 

В приводимом примере рассматривается строка «Роза увяла от мороза», в которой 
ищется подстрока «роза». Если учитывать регистр, то такая подстрока одна, без 
учета регистра их две. 

begin // p06022 

  var s := 'Роза увяла от мороза'; 

  s.IsMatch('роза').Println; // True 

  s.IsMatch('роза\s').Println; // False 

  s.IsMatch('Роза\s').Println; // True 

  s.IsMatch('роза\s', RegexOptions.IgnoreCase).Println; // True 

  s.IsMatch('(?i)роза').Println; // True 

end. 

В предпоследнем варианте проверки опция RegexOptions.IgnoreCase означает 
требование игнорировать регистр букв. В последнем варианте тот же эффект 
достигается при помощи конструкции (?i). 

6.2.19.3 RegexReplace – замена подстрок в строке 
Расширение s.RegexReplace(reg, ss, opt) возвращает строку, полученную заменами в 
строке s подстроки, удовлетворяющей регулярному выражению, подстрокой ss. С 
помощью opt можно задавать дополнительные опции. Делаются замены всех 
найденных подстрок. 

begin // p06023 

  var s := 'Мама мыла раму, Маша ела кашу'; 

  s.RegexReplace('М','Д').Println; // Дама мыла раму, Даша ела кашу 

  s:='23*x-Sin(x)'; 

  s.RegexReplace('x','(x+1.5)').Println; // 23*(x+1.5)-Sin((x+1.5)) 

  s.RegexReplace('\-Sin\(x\)','').Println; // 23*x 

end. 

В последней замене, показывающей удаление подстроки, обратите внимание на 
экранирование. Начинающим полезно вначале убедиться при помощи IsMatch в 
правильности поиска, а затем делать замену. 

Вторая форма расширения s.RegexReplace(reg, Match -> string, opt) отличается тем, 
что найденные подстроки заменяются их преобразованием, заданным лямбда-
выражением. Match – это объект одноименного класса, являющийся результатом 
каждого удачного сопоставления шаблона reg со строкой s.  

begin // p06024 

  var s := 'Мама мыла раму'; 

  // смена регистра 

  s.RegexReplace('.а', m -> UpperCase(m.Value)).Println; // МАМА мыЛА РАму 

  // номер позиции вхождения подстроки 

  s.RegexReplace('.а', m -> m.Index.ToString).Println; // 02 мы7 10му 

  // длины слов 

  s := 'А роза упала на лапу Азора'; 

  s.RegexReplace('\w+', m -> m.Length.ToString).Println; // 1 4 5 2 4 5 

  // заменить пробелы их количеством 

  s := '       тестовая  строчка   '; 

  s.RegexReplace('\s+', m -> m.Length.ToString).Println; // 7тестовая2строчка3 

end. 

6.2.19.4 MatchValue – поиск первого вхождения подстроки 
Расширение s.MatchValue(reg, opt) возвращает первую из подстрок в строке s, 
соответствующую регулярному выражению reg. С помощью opt можно задавать 
дополнительные опции. В отличии от расширения IsMatch (глава 6.2.19.2) возвра-
щает подстроку, а не логическое значение, показывающее успешность поиска. В 
качестве примера возьмем тот же самый, что для IsMatch. 

begin // p06025 

  var s := 'Роза увяла от мороза'; 

  s.MatchValue('роза').Println; // роза 

  s.MatchValue('роза\s').Println; // пустая строка 

  s.MatchValue('Роза\s').Println; // Роза 

  s.MatchValue('роза\s', RegexOptions.IgnoreCase).Println; // Роза 

  s.MatchValue('(?i)роза').Println; // Роза 

end. 

6.2.19.5 MatchValues – поиск всех вхождений подстроки 
Расширение s.MatchValues(reg, opt) возвращает последовательность подстрок в 
строке s, соответствующую регулярному выражению reg. С помощью opt можно 
задавать дополнительные опции. 

begin // p06026 

  var s := 'Роза увяла от мороза'; 

  s.MatchValues('роза').Println; // роза 

  s.MatchValues('роза\s').Println; // пустая строка 

  s.MatchValues('Роза\s').Println; // Роза 

  s.MatchValues('роза\s', RegexOptions.IgnoreCase).Println; // Роза 

  s.MatchValues('(?i)роза').Println; // Роза роза 

end. 

В последнем обращении к MatchValues было найдено две подстроки. Первая полу-
чена как слово «Роза» с игнорированием регистра, вторая – как подстрока слова 
«мороза». 

6.2.19.6 Matches – результаты поиска в элементах типа Match 
Расширение s.Matches(reg, opt) возвращает последовательность элементов типа 
Match из строки s, соответствующих регулярному выражению reg (подраздел 
6.2.19). С помощью opt можно задавать дополнительные опции. 

Тип Match объявлен в пакете System.Text.RegularExpressions и подробная справка 
по нему может быть найдена в описании этого пакета. Каждый элемент Match 
хранит всю информацию относительно результатов поиска, в том числе, саму 
найденную подстроку, ее длину и позицию в исходной строке. Некоторые детали 
приводятся в подразделе 6.2.19.8. 

Пример работы с расширением Matches был дан в разделе 6.2.18. 

6.2.19.7 Опции регулярного выражения 
Опции (необязательные возможности) RegexOprions регулярных выражений в 
пределах раздела 6.2.19 обозначались как opt. Приведем лишь наиболее употреби-
тельные четыре опции из десяти имеющихся них. В скобках указаны коды опций. 
 None (0) – указывает на отсутствие опций; 
 IgnoreCase (1) – игнорирование регистра буквенных символов; 
 Multiline (2) – многострочный режим. Если текст содержит символы перевода 

строки, они рассматриваются, как обычные символы и поиск производится во 
всей строке; 

 Singleline (16) – однострочный режим. Если текст содержит символы перевода 
строки, строка рассматривается только до первого такого символа. 

Опции задаются в виде параметра RegexOptions.Опция. Вместо имен опций можно 
писать их коды в виде RegexOptions(Код). Если нужно указать несколько опций, 
коды складываются, как целые числа. 

Некоторый недостаток использования опций состоит в том, что они действуют в 
пределах всего регулярного выражения. В отдельных случаях может помочь ис-
пользование опций непосредственно в выражении. С этой целью в нужных местах 
включаются конструкции вида (?⍟) или (?-⍟), где ⍟ - так называемый инлайно-
вый символ, обозначающий опцию. В пакете System.Text.RegularExpressions име-
ются пять инлайновых символов, но мы отметим три из них: 
 i – IgnoreCase (1); 
 m – Multiline (2); 
 s – Singleline (16). 

Знак «минус» перед инлайновым символом означает отмену действия соответ-
ствующей опции. 

По умолчанию действует многострочный режим с учетом регистра.   

begin // p06027 

  var s := 'Карл у Клары украл кораллы'; 

  // Карл Клары крал кораллы 

  s.MatchValues('(?i)к\w*(?-i)(ар|ра)\w*').Println; 

  s.MatchValues('[Кк]\w*(ар|ра)\w*').Println; 

end. 

Первый шаблон приведен лишь для того, чтобы показать, как используются ин-
лайновые символы. Второй шаблон показывает, что в данном случае можно пре-
восходно обойтись и без них. 

6.2.19.8 Элементы Matches 
Как уже упоминалось, тип Match объявлен в пакете System.Text.RegularExpressions. 
Каждая подстрока, соответствующая успешному поиску в соответствии с шабло-
ном, сохраняется в объекте типа Match. Объект имеет достаточно сложную струк-
туру, но на практике чаще всего используются только три его поля: 

 Match.Value – сохраненная подстрока; 
 Match.Index – позиция в исходной строке, в которой найден первый символ 

сохраненной подстроки (отсчет с единицы); 
 Match.Length – длина сохраненной подстроки. 

Практически, расширение MatchValue возвращает в качестве результата значение 
поля Value первого из элементов Matches, а MatchValues – последовательность 
значений поля Value из всех элементов Matches. 

6.2.20 Преобразование строки в динамический массив 
Строку логически можно рассматривать, как динамический массив символов. Но 
можно действительно преобразовать строку в динамический массив символов. 
 s.ToCharArray – метод, возвращающий динамический массив типа char, полу-

ченный путем посимвольного разбиения исходной строки s. Обратное преобра-
зование из массива или последовательности символов (массив превращается в 
последовательность в подавляющем количестве операций с ним – это вы уже 
знаете) в строку можно выполнить при помощи расширения .JoinIntoString. 

Пусть требуется выбрать из заданной строки все символы по одному разу, отсор-
тировать их по возрастанию и заменить полученной строкой исходную. 

begin // p06028 

  var s := 'наша тестовая строка'; 

  s := s.ToCharArray.Distinct.Sorted.JoinIntoString; 

  s.Println //  авекнорстшя 

end. 

6.2.21 Преобразование целых чисел к строке 
Целые числа может потребоваться преобразовать к строковому представлению 
для формирования строк вывода, для анализа цифр в числе (число цифр, одинако-
вые цифры, перестановки цифр и т.д.), для передачи куда-либо данных в символь-
ном виде и во многих других случаях. 
 Str(n, s) – процедура. Преобразует значение n любого целочисленного типа, 

кроме BigInteger, к строке s; 
 IntToStr(n) – функция. Возвращает преобразованное к строке значение n типов 

integer и int64;  
 ⍟ n.ToString – метод. Возвращает преобразованное к строке значение n. 

Метод .ToString является единственным из перечисленных, позволяющим преоб-
разовать к строке тип BigInteger. 

6.2.22 Преобразование вещественных чисел к строке 
Вещественные числа преобразуют к строке в тех же случаях, что и целые. 
 Str(n, s) – процедура. Преобразует значение n типа real или single к строке s; 
 FloatToStr(n) – функция. Возвращает преобразованное к строке значение n типа 

real;  
 n.ToString – метод. Возвращает преобразованное к строке значение n веще-

ственного типа. 

6.2.23 Преобразование строки к числу 
С преобразованием числа к строке все достаточно просто. Обратное преобразова-
ние намного сложнее. 

Строка может целиком представлять одно число. В этом случае говорят о преобра-
зовании строки к числу. Строка может содержать представление нескольких чисел 
одного типа, отделенных друг от друга каким-то разделителем или группой разде-
лителей (чаще всего – пробелом, знаком табуляции #09 или запятой). Такая строка 
может быть преобразована к массиву чисел. Наконец, во всех случаях строка может 
содержать некорректное представление числа или одного из ряда чисел. 

6.2.23.1 Преобразование строки к целому числу 
 s.ToInteger – расширение. Возвращает содержимое сроки s, преобразованное к 

значению типа integer. В случае ошибке генерируется исключение; 
 StrToInt(s) – функция. Возвращает содержимое сроки s, преобразованное к 

значению типа integer. В случае ошибке генерируется исключение; 
 s.ToInteger(k) – расширение. Возвращает содержимое сроки s, преобразованное 

к значению типа integer. В случае ошибки возвращается значение k типа integer; 
 Val(s, n, err) – процедура. Преобразует содержимое строки s к значению n типа 

integer. В случае успешного преобразования err=0, в противном случае err>0; 
 TryStrToInt(s, n) – функция. Преобразует содержимое строки s к значению n 

типа integer. В случае успешного преобразования возвращает True. Если преоб-
разование не удалось, возвращает False; 

 s.TryToInteger(n) – расширение. Преобразует содержимое строки s к значению n 
типа integer. В случае успешного преобразования возвращает True. Если преоб-
разование не удалось, возвращает False; 

 integer.TryParse(s) – статический метод. Возвращает содержимое сроки s, пре-
образованное к значению типа integer. В случае успешного преобразования воз-
вращает True. Если преобразование не удалось, возвращает False; 

 StrToInt64(s) – функция. Возвращает содержимое сроки s, преобразованное к 
значению типа int64. В случае ошибке генерируется исключение; 

 TryStrToInt(s, n) – функция. Преобразует содержимое строки s к значению n 
типа int64. В случае успешного преобразования возвращает True. Если преобра-
зование не удалось, возвращает False; 

 int64.TryParse(s) – статический метод. Возвращает содержимое сроки s, преоб-
разованное к значению типа int64. В случае успешного преобразования возвра-
щает True. Если преобразование не удалось, возвращает False; 

 s.ToBigInteger – Возвращает содержимое сроки s, преобразованное к значению 
типа BigInteger. В случае ошибке генерируется исключение;. 

Статический метод .TryParse также имеется в целочисленных классах shortint, 
byte, smallint, word, cardinal и BigInteger. 

6.2.23.2 Преобразование строки к вещественному числу 
 s.ToReal – расширение. Возвращает содержимое сроки s, преобразованное к 

значению типа real. В случае ошибке генерируется исключение; 
 StrToReal(s) – функция. Возвращает содержимое сроки s, преобразованное к 

значению типа real. В случае ошибке генерируется исключение; 
 StrToFloat(s) – функция. Возвращает содержимое сроки s, преобразованное к 

значению типа real или single. В случае ошибке генерируется исключение; 
 s.ToReal(r) – расширение. Возвращает содержимое сроки s, преобразованное к 

значению типа real. В случае ошибки возвращается значение r типа real; 
 TryStrToReal(s, r) – функция. Преобразует содержимое строки s к значению r 

типа real. В случае успешного преобразования возвращает True. Если преобра-
зование не удалось, возвращает False; 

 TryStrToFloat(s, r) – функция. Преобразует содержимое строки s к значению r 
типа real или single. В случае успешного преобразования возвращает True. Если 
преобразование не удалось, возвращает False; 

 TryStrToSingle(s, r) – функция. Преобразует содержимое строки s к значению r 
типа single. В случае успешного преобразования возвращает True. Если преобра-
зование не удалось, возвращает False 

 s.TryToReal(r) – расширение. Преобразует содержимое строки s к значению r 
типа real. В случае успешного преобразования возвращает True. Если преобра-
зование не удалось, возвращает False; 

 real.TryParse(s) – статический метод. Возвращает содержимое сроки s, преобра-
зованное к значению типа real. В случае успешного преобразования возвращает 
True. Если преобразование не удалось, возвращает False. 

Статический метод .TryParse также имеется в классе single. 

6.2.23.3 Преобразование строки к массиву чисел 
 s.ToIntegers – расширение. Возвращает содержимое сроки s, преобразованное к 

динамическому массиву значений типа integer. В случае ошибке генерируется 
исключение; 

 s.ToReals – расширение.  Возвращает содержимое сроки s, преобразованное к 
динамическому массиву значений типа real. В случае ошибке генерируется ис-
ключение. 

Разделителями могут выступать только пробельные символы. 

6.2.24 Пример с преобразованиями строк и чисел 
Пусть дана некоторая строка. Внутри строки содержатся простые выражения вида 
«(Число ЗнакОперации Число)», где числа только целые без знака, знак операции – 
это «+», «-», «*» и «/». Числа и знаки операции зашумлены прочими символами. 
Требуется выделить выражения и каждое записать на отдельной строке в виде 
Выражение = ВычисленноеЗначение. 

function Calc(v: string): string; // p06029 

begin 

  var m := v.MatchValues('(\d+|[+\-*/])').ToArray; 

  var a := m[0].ToInteger; 

  var op := m[1][1]; 

  var b := m[2].ToInteger; 

  case op of 

    '+': Result := ' = ' + (a + b); 

    '-': Result := ' = ' + (a - b); 

    '*': Result := ' = ' + (a * b); 

    '/': Result := ' = ' + (a / b); 

  end; 

  Result := v + Result 

end; 

 

begin 

  var s := 'ar(f?3yh535v^+et14!f)re(@2T6\s/1we1r)(y14 2p*23)gr(18-253)'; 

  s.RegexReplace('[^()+\-*/0-9]', '').MatchValues('\([^)]+\)') 

      .Select(v -> Calc(v)).PrintLines 

end. 

Будут получены следующие результаты: 

(3535+14) = 3549 

(26/11) = 2.36363636363636 

(142*23) = 3266 

(18-253) = -235 

Замена с помощью RegexReplace удалит из строки весь «мусор». Использован шаб-
лон '[^()+\-*/0-9]', который заменяет пустой строкой (а по факту просто удаляет) 
подстроки-символы, не являющиеся круглыми скобками, знаками арифметических 
операций и цифрами. Как обычно, символ «минус» экранирован, поскольку его 
наличие в символьном классе создает диапазон. Полученная строка передается 
MatchValues c шаблоном, формирующим последовательность подстрок, состоящих 
из элементов, заключенных в круглые скобки. Подстроки проецируются на функ-
цию Calc, делающую вычисления и формирующую окончательный вид подстроки. 
Полученная последовательность подстрок отображается на мониторе посредством 
PrintLines. 

Функция Calc получает из входной строки вида «(Число1 ЗнакОперации Число2)»  
помощью шаблона в MatchValues последовательность из трех элементов типа 
string. Поскольку последовательность не хранится, а с каждым элементом нужно 
работать отдельно, последовательность сохраняется в массиве m. Теперь подстро-



214 | Часть 6   
 

ка представлена элементом m[0] для Число1, m[1] для знака операции и m[2] для 
Число2. Поместим числовые значения для Число1 и Число2 в переменные a b и 
соответственно. 

Операция – это строка длиной в один символ. И вот здесь есть неудобство, зало-
женное идеологией языка Паскаль. Тип односимвольного литерала нельзя интер-
претировать однозначно. В операторе case используются как раз односимвольные 
литералы. И компилятор интерпретирует их тип, как char. Массив m имеет тип 
array of string, поэтому m[1] тоже имеет тип string. При попытке использовать 
m[1] в case компилятор зафиксирует несоответствие типа. В Паскале нет отдель-
ных функций или методов для преобразования типа string -> char и наоборот. 
Поэтому приходится прибегать к шаманству. Если s – строка типа string, то ее 
отдельный символ будет иметь тип char. Поэтому мы можем написать s[1] и полу-
чить значение нужного нам типа. Преобразовать символ «с» к строке проще: можно 
его присвоить переменной типа string или сложить с пустой строкой: с + ''. 

Вычисленное значение автоматически преобразуется к строке, поскольку таковы 
правила преобразования в операции «+» соединения строки с числом (раздел 
6.2.4). 

6.2.25 Чтение данных из строки 
Операция чтения данных из строки s состоит в получении подстрок и преобразо-
вании их в значения требуемого типа. Указывается начальная позиция чтения 
from, а после успешного чтения from увеличивается на длину прочитанной под-
строки, позволяя считывать очередное значение. 

Чтение данных может быть полезно при получении строк из текстовых файлов 
(часть 9). Предполагается, что числа в строке могут быть разделены произвольным 
количеством любых пробельных символов, в том числе, признаками смены строки. 
Это позволяет прочитать в одном цикле многострочный файл. 

6.2.25.1 Чтение из строки данных типа integer 
 s.ReadInteger(from) – расширение. Начиная с позиции from, считывается сим-

вольное представление целого числа, ограниченного пробельным символом 
или концом строки. Затем оно преобразуется и возвращается как значение типа 
integer. Значение from увеличивается на длину прочитанной подстроки и будет 
содержать позицию первого после окончания числа пробельного символа. Если 
преобразование не удалось или from указано за пределами строки s, генериру-
ется исключение; 

 ReadIntegerFromString(s, from) – функция. Делает то же самое; 
 TryReadIntegerFromString(s, from, n) – функция. Начиная с позиции from, считы-

вается символьное представление целого числа и делается попытка преобразо-
вать его к целому числу. Если попытка успешна, значение сохраняется в n и 
функция возвращает True. В противном случае функция возвращает False. 

begin // p06030 

  var s := '123'#9#13#10'56 00001'; // так тоже можно строить строки 

  var from := 1; 

  loop 3 do 

    s.ReadInteger(from).Print // 123 56 1 

end. 

6.2.25.2 Чтение из строки данных типа real 
 s.ReadReal(from) – расширение. Начиная с позиции from, считывается символь-

ное представление вещественного числа, ограниченного пробельным символом 
или концом строки. Затем оно преобразуется и возвращается как значение типа 
real. Значение from увеличивается на длину прочитанной подстроки и будет со-
держать позицию первого после окончания числа пробельного символа. Если 
преобразование не удалось или from указано за пределами строки s, генериру-
ется исключение; 

 ReadRealFromString(s, from) – функция. Делает то же самое; 
 TryReadRealFromString(s, from, r) – функция. Начиная с позиции from, считыва-

ется символьное представление вещественного числа и делается попытка пре-
образовать его к вещественному числу. Если попытка успешна, значение сохра-
няется в r и функция возвращает True. В противном случае функция возвращает 
False. 

begin // p06031 

  var s := 'Значение 3.1416 приблизительно равно числу "пи"'; 

  var from := Pos('3', s); 

  Println('Считали число', s.ReadReal(from)) // Считали число 3.1416 

end. 

6.2.25.3 Чтение из строки символьных данных 
 s.ReadWord(from) – расширение. Начиная с позиции from, считывается слово, 

ограниченное пробельным символом или концом строки. Возвращаемое значе-
ние имеет тип string. Значение from увеличивается на длину прочитанного сло-
ва будет содержать позицию первого после окончания слова пробельного сим-
вола. При попытке читать за пределами строки генерируется исключение; 

 ReadWordFromString(s, from) – функция. Делает то же самое. 

Вот так можно вывести строку, удалив из нее лишние пробелы: 

begin // p06032 

  var s := '   Карл   у   Клары украл   кораллы'; 

  var (l, from) := (s.Length, 1); 

  while from <= l do 

    Print(s.ReadWord(from)); // Карл у Клары украл кораллы  

  Println 

end. 

6.2.26 Форматирование данных для вывода 
Для оформления выводимых данных в языке Паскаль традиционно используется 
процедуры Write и Writeln, в которых для выводимого элемента данных может 
указываться минимальная ширина поля вывода w, а для вещественных значений – 
еще и количество цифр в дробной части t. Разделителем служит символ «двоето-
чие» (:). Если параметр w не указан или имеет недостаточную для отображения 
выводимого значения величину, значение выводится со своей фактической дли-
ной. Если для вещественного числа параметр t не указан, в дробной части выво-
дится фактическое количество цифр, но при этом общее количество цифр не может 
превышать 15. Если в дробной части больше цифр, чем указано в t, выводится t 
цифр и при этом делается округление. При указании значения t, требующего в 
выводимом числе отобразить более 15 цифр, все цифры после пятнадцатой будут 
представлены нулями. 

begin // p06033 

  var (a, b) := (-23.4723423290834, 16554); 

  Writeln(a, ' ', b, ' ', a * b); 

  Writeln(a, ' ', b, ' ', a * b:0:5); 

  Writeln(a, ' ', b, ' ', a * b:20); 

  Writeln(a, ' ', b, ' ', a * b:25:5); 

  Writeln(a, ' ', b, ' ', a * b:0:20); 

  Writeln('Тестовая строка', ' ', '*'); 

  Writeln('Тестовая строка':10, ' ', '*'); 

  Writeln('Тестовая строка':30, ' ', '*'); 

end. 

Обратите внимание на частое использование нуля при задании ширины поля 
вывода w. Это типичный прием вывода вещественных значений в Паскале, когда 
неизвестно количество цифр в целой части числа. 

-23.4723423290834 16554 -388561.154915647 

-23.4723423290834 16554 -388561.15492 

-23.4723423290834 16554    -388561.154915647 

-23.4723423290834 16554             -388561.15492 

-23.4723423290834 16554 -388561.15491564700000000000 

Тестовая строка * 

Тестовая строка * 

               Тестовая строка * 

В случает, когда значение вещественного числа слишком мало или велико, будет 
использован формат вывода с плавающей точкой вида x.x…xЕ∓nn (x.x…x∙10∓nn). 

Выводимые числовые значения прижимаются к правой границе отведенного 
для вывода поля шириной w за счет добавления слева нужного количества 
пробелов. Это общепринято. Точно так же, принято прижимать строки к левой 
границе поля, поэтому они дополняются пробелами справа. Но тут Паскаль 
ведет себя странно: пробелы все равно добавляются слева и об этом надо пом-
нить. 

Существуют и другие способы вывода данных. Их общая идея состоит в том, чтобы 
формировать строки такими, какими они должны отображаться и затем выводить 
без изменения. 

6.2.26.1 Выравнивание строки пробелами 
Чтобы строка s длиной L имела нужную ширину w, можно дополнить ее w-L пробе-
лами слева или справа. Например, вот так: 

s := ’ ‘ * (w - s.Length) + s; 

s := s + ’ ‘ * (w - s.Length); 

Но так писать для каждого выводимого элемента неудобно. Поэтому в классе 
string были введены два метода: 
 s.PadRight(w) – возвращает строку s, дополненную пробелами справа до длины 

w; 
 s.PadLeft(w) – возвращает строку s, дополненную пробелами слева до длины w. 

6.2.26.2 Составное форматирование 
Выравнивание строки пробелами решает проблему с выводом символьных дан-
ных. Для любого типа данных есть общее решение – использование составного 
форматирования (composite format в терминологии Microsoft .NET Framework). 

Составное форматирование использует в качестве исходных данных строку со-
ставного формата (шаблон, задающий правила преобразования данных) и список 
объектов, подлежащих форматированию. Этот список следует за строкой составно-
го формата и отделяется от нее запятой. Элементы в списке нумеруются от нуля и 
разделяются запятыми. 

Строка составного формата может содержать произвольную комбинацию некото-
рого количества неизменяемого текста, представленного литералами, и элементов 
форматирования, каждый из которых заключается в фигурные скобки. 

Элемент форматирования является местозаполнителем (поле, в котором впо-
следствии будет размещено значение) и в простейшем случае содержит только 
индекс – порядковый номер элемента, подлежащего форматированию, в списке. 

begin 

  Format('Длина окружности диаметра {0} равна {1}', 2.16, 2.16 * Pi).Println 

end. 

Длина окружности диаметра 2.16 равна 6.78584013175395 

В общем случае элемент форматирования имеет вид 

{индекс, выравнивание : формат}, 

где выравнивание – целое значение со знаком, указывающее общую ширину от-
форматированного поля, формат – строка формата для элемента. 

Если выравнивание имеет отрицательное значение, результат прижимается влево 
путем добавления справа пробелов до нужной ширины. 

В Microsoft .NET Framework определен обширный набор форматов и здесь будет 
рассмотрена только небольшая их часть. В частности, не рассматриваются настра-
иваемые числовые форматы. 

§ 1. Для целочисленных данных используется формат d или D (decimal), за которым 
может быть указано желаемое количество цифр. 

begin 

  Format('*{0,20:d}*{0,-15:D6}*{0:d1}*', -152).Println 

end. 

*                -152*-000152        *-152* 

Здесь трижды указан индекс 0, что позволило три раза использовать одно и то же 
значение -152 из списка. Числовое значение после d (D) определяет минимальное 
желаемое количество выводимых цифр, значение выравнивания перед двоеточием 
– общее количество позиций. Обратите внимание на появление незначащих нулей. 

§ 2. Формат вещественных данных f или F определяет преобразование для отобра-
жения с фиксированной точкой (fixed), т.е. в виде целой и дробной части, разде-
ленных точкой. По умолчанию в дробной части будут указаны две цифры с округ-
лением значения. 

begin 

  Format('*{0,20:f}*{0,-15:f6}*{0:f1}*', -152.327).Println 

end. 

*             -152.33*-152.327000    *-152.3* 

§ 3. Формат вещественных данных e или E определяет преобразование для отоб-
ражения с плавающей точкой (exponential), т.е. в виде одной цифры в целой части, 
нужного количества цифр в дробной части и показателя степени числа 10, отде-
ленного буквой E. Этот формат обычно используется для величин, имеющих очень 
малые или очень большие значения. По умолчанию число знаков в дробной части 
равно шести. 

begin 

  Format('*{0,20:e}*{0,-15:E6}*{0:e1}*', -152.327e-31).Println 

end. 

*      -1.523270e-029*-1.523270E-029 *-1.5e-029* 

§ 4. Общий формат числовых данных g или G определяет преобразование для 
данных любого числового типа. Наиболее подходящий формат (d, f, e) компилятор 
выбирает самостоятельно. Для отображения значений real с максимальной точно-
стью используется формат g17, для single – g9. 

begin 

  Format('*{0,20:g}*{0,-15:G6}*{0:g1}*', -152).Println; 

  Format('*{0,20:g}*{0,-15:g6}*{0:g1}*', -152.327).Println; 

  Format('*{0,20:g}*{0,-15:G6}*{0:g1}*', -152.327e-31).Println 

end. 

*                -152*-152           *-2e+02* 

*            -152.327*-152.327       *-2e+02* 

*        -1.52327e-29*-1.52327E-29   *-2e-29* 

§ 5. Формат для преобразования целочисленных данных к их шестнадцатеричному 
представлению x или X. По умолчанию отображается минимально-необходимое 
количество цифр. 

begin 

  Format('*{0,20:x}*{0,-15:X6}*{0:x1}*', 1023).Println; 

  Format('*{0,20:x}*{0,-15:X6}*{0:x1}*', -1023).Println 

end. 

*                 3ff*0003FF         *3ff* 

*            fffffc01*FFFFFC01       *fffffc01* 

6.2.26.3 Функция и статический метод Format 
Составное форматирование является основой преобразования строк при помощи 
функции и статического метода Format. 
 Format(cf, params) – функция. Возвращает строку,  полученную форматировани-

ем списка параметров params при помощи строки составного формата cf; 
 string.Format(cf, params) – статический метод. Делает то же самое. 

Примеры работы с Format приведены в предыдущей главе. 

6.2.26.4 Вывод с использованием составного форматирования 
 WriteFormat(сf, params) – процедура. Форматирует список параметров params 

при помощи строки составного формата cf и выводит полученную строку; 
 WritelnFormat(сf, params) – процедура. Делает то же самое, осуществляя затем 

переход на новую строку вывода. 

6.2.26.5 Интерполированные строки 
Интерполированная строка – это объединение строки составного формата и спис-
ка объектов, подлежащих преобразованию в строковое представление. Получается 
как бы «два в одном» в сравнении с компонентами Format. 

Интерполированная строка представляет собой литерал, перед которым записан 
символ $. Литерал содержит интерполированные выражения. Интерполированное 
выражение очень похоже на элемент форматирования строки составного формата: 
оно так же заключается в фигурные скобки и содержит описание формата данных, 
только вместо индекса подлежащего форматированию выражения указывается 

само это выражение. Проще всего это продемонстрировать в сравнении с приме-
рами из главы 6.2.26.2. 

begin // p06034 

  Format('Длина окружности диаметра {0} равна {1}', 2.16, 2.16 * Pi).Println; 

  $'Длина окружности диаметра {2.16} равна {2.16 * Pi}'.Println; 

  Format('*{0,20:d}*{0,-15:D6}*{0:d1}*', -152).Println; 

  $'*{-152,20:d}*{-152,-15:D6}*{-152:d1}*'.Println; 

  Format('*{0,20:f}*{0,-15:f6}*{0:f1}*', -152.327).Println; 

  $'*{-152.327,20:f}*{-152.327,-15:f6}*{-152.327:f1}*'.Println 

end. 

Обычно интерполированные строки получаются короче и нагляднее, чем строки 
составного форматирования. 

begin 

  var (a, b) := (5.2, 4.17); 

  Format('Сумма квадратов катетов a={0} и b={1} равна {2}', 

      a, b, a * a + b * b).Println; 

   

 

$'Сумма квадратов катетов a={a} и b={b} равна {a * a + b * b}'.Println 

end. 

Сумма квадратов катетов a=5.2 и b=4.17 равна 44.4289 

Сумма квадратов катетов a=5.2 и b=4.17 равна 44.4289 

Интерполированная строка – это строка типа string и с ней можно работать обыч-
ным образом. 

begin // p06035 

  var (a, b) := (5.2, 4.17); 

  var s1 := $'Сумма квадратов катетов a={a} и b={b} равна {a*a+b*b}'; 

  var s2 := $'{a*a} + {b*b} = {a*a+b*b}'; 

  s2.Println; 

  s1.Println 

end. 

27.04 + 17.3889 = 44.4289 

Сумма квадратов катетов a=5.2 и b=4.17 равна 44.4289 


